NoBLE for percolation

Implementation of the computer-assisted proof of the NoBLE by Robert Fitzner and Remco van der Hofstad

Robert Fitzner
Eindhoven University of Technology

Abstract

This document is the second part of the computer-assisted proof of the non-backtracking lace expansion (NoBLE). The NoBLE was created by Remco van der Hofstad and the author to prove mean-field behavior for self-avoiding walk, lattice tree (LT), lattice animals (LA) and percolation. In this file the computations for percolation are performed. The technique is explained in the paper “Generalized approach to the non-backtracking lace expansion” (which we will consistently refer to as (I)), and the bounds that we implement here are derived in “Nearest-neighbor percolation function is continuous for $d > 10$" (which we will refer to as (II)). All references in this file are to result and equations in one of these two papers.

This file is accompanied by the notebooks ***SRW.nb*** and ***General.nb***. In the SRW files a number of simple random walks quantities are computed. In ***General.nb*** general bounds, derived in (I), are implemented. Before doing computations with this file the user should first open these files, choose a dimension and execute all lines of these files.

Then, the user is expected to choose constants Γ_i and c_μ, $c_{n,l,S}$ in this file. Examples for these constants that work in $d=11$ are provided below. After choosing these quantities the user should select the menu item Evaluate-> Evaluate Notebook. In a table at the end of this document the results of the computations are shown: it can be seen whether the bootstrap argument, with the given parameters, and therefore the analysis, is succesful.

The computation of the ***SRW*** and ***General*** file are independent of the values Γ_i and c, so that we need to execute these files only once, when you start the mathematica Kernel or when you want to change the dimension under consideration.

3.11.2018

Input

Here the user should choose the constants Γ_i, c_μ and $c_{n,l,S}$ for which we try to perform the bootstrap, explained in (I), using this program/document. These values are introduced in Section 2.1, see (2.1)-(2.3) and coments below. For the set S used to define f_3, we are restricted to the sets $\{0\}$ and $\mathbb{Z}^d\backslash\{0\}$. For the following values the bootstrap succeeds in dimension 11:

In[4929]:=
```
(*The parameter choice that works in d=11*)
Gamma1 = 1.01306; (*(2d-1)p*)
Gamma2 = 1.07513; (*(2d-1)/(2d-2) sup_k[1-D̂(k)] τ̂_p(k)*)
Gamma3 = 1;
cmu = 1.0029684;
c[1, 6, 0] = 0.008714; (*Assumed bound on ∑_y |y|_2^2 τ(y) (D^{*6}*τ*τ) (-y)*)
c[0, 0, 1] = 0.0661; (*Assumed bound on sup_{x≠0} |x|_2^2 τ(x)*)
c[1, 0, 1] = 0.108; (*Assumed bound on sup_{x≠0} ∑_y |y|_2^2 τ(y) τ(x-y)*)
c[1, 1, 1] = 0.05352; (*Assumed bound on sup_{x≠0} ∑_y |y|_2^2 τ(y) (D*τ) (x-y)*)
c[1, 2, 1] = 0.0404; (*Assumed bound on sup_{x≠0} ∑_y |y|_2^2 τ(y) (D*D*τ) (x-y)*)
c[1, 3, 1] = 0.022 ;
(*Assumed bound on sup_{x≠0} ∑_y |y|_2^2 τ(y) (D^{*3}*τ) (x-y)*)
```

Let us explain how we found those values: We with the bound $f_i(z_I)$ at z_I, which is indepented of the improvement of bound arguments and give a good orientation. Starting from these values we used a semi-automated procedure to find appropriate values for these constants. The basic idea is based on a fixed point idea and works as follow: Starting from some reasonable values we conclude bounds. Then, we use these bounds as new assumed bound and re-do all computations again starting from theses values. We repeat this procedure until either values of the concluded value start to diverge and we failed or until we actually concluded bounds that were smaller than the bounds assumed in this run. This is implemented near the end of this document.

Sandbox

To improve readablity of the code, comments are added. These comments are found between the signs (* comments *). We prepare the following input field in case you want to try other values of the constants. For this, you can just remove the commenting (* / *) in the following input field:

In[4932]:=
```
(*Gamma1=1.01306;
Gamma2=1.07513;
Gamma3=1;
cmu=1.0029684;
c[1,6,0]=0.008714;(* c_{1,6,{0}}*)
c[0,0,1]=0.0661;(* c_{0,0,Z/{0}}*)
c[1,0,1]=0.108;(* c_{1,0,Z/{0}}*)
c[1,1,1]=0.05352;(* c_{1,1,Z/{0}}*)
c[1,2,1]=0.0404;(* c_{1,2,Z/{0}}*)
c[1,3,1]=0.022;
(* c_{1,3,Z/{0}}*)*)
```

Bound on the two-point function and on repulsive diagrams

In this document we use the notation of (I). This means that we use z for the probability that an edge is occupied and $G_{m,z}$ for the two-point function (instead of $\tau_{m,p}$ as used in (I)).

Definition of constants

We define the constants for two settings: we use s=i for the bounds on the inital value of the bootstrap $z = z_I$ and s=o for improvement of the bounds for $z \in (z_I, z_c)$.

```
In[4933]:= z[i] = 1/(2 d - 1);
z[o] = Gamma1/(2 d - 1);(* Upper bound on z ∈ (z_I, z_c) and thereby also on z_c *)
(*bound on the two-point function G(k)< Vargamma2 C_{1/2d}(x),
for i: z=z_i, Ĝ(k)< Vargamma2 Ĉ_{1/2d}(k), for o: z in (z_i, z_c)*)
VarGamma2[i] = (2 d - 2)/(2 d - 1);
VarGamma2[o] = Gamma2 * (2 d - 2)/(2 d - 1);
```

Bounds on two-point functions

Here, we compute bounds on $G_{m,z}(x)$ for some explicit values of x. These bounds are obtained by extracting short, explicit contributions and by bounding the longer contributions using f_2, see Section 5.3 of (I):

$$G_{m,z}(x) \leq \sum_{j=m}^{M-1} z^j c_j(x) + G_{M,z}(x) \leq \sum_{j=m}^{M-1} \left(\frac{\Gamma_1}{2d-1}\right)^j c_j(x) + \left(\frac{\Gamma_1}{2d-1}\right)^M \frac{2d-2}{2d-1} \Gamma_2 I_{1,M}(0)$$

for even M. To bound the short explicit contributions we use the number of j-step self-avoiding walks ending at x, $c_j(x)$ = nrSAW[j,d,x], provided in the model-independent SRW-integral notebook, these are provided for $j \leq$ *ComputedSteps*. The extraction of short contributions creates a better bound than just applying f_2. The reason is that f_1 gives a much sharper bound than f_2.

```
In[4937]:= (*Number of steps we will extract,
we have to choose M=Rsteps even to apply the bound.*)
If[EvenQ[ComputedSteps],
 Explicit = ComputedSteps;
 (*the length of the polygon that we extract. It is supposed to be even*)
 RSteps = ComputedSteps + 2;,
 (*The length of the shortest loop in the remainder terms*)
 Explicit = ComputedSteps - 1;
 RSteps = ComputedSteps + 1;]
```

Bound on $G_{m,z}(e_1)$

We use that the variable *Explicit* is even and that $G_{m,z}(e_1) = (D * G_{m,z})(0)$ to compute

```
In[4938]:= Do[
 Do[
  Bound[G, {1}, m, s] = Sum[z[s]^j nrSAW[j, d, {1}], {j, m, Explicit}] +
     (2 d z[s])^(RSteps-1) VarGamma2[s] Ivalue[1, RSteps, {0}];
  , {m, 1, 5}]
 , {s, {i, o}}]
```

Bound on $G_{m,z}(e_1 + e_2)$

We use $G_{m,z}(e_1 + e_2) \leq \frac{d}{d-1}\left(D^2 * G_{m,z}\right)(0)$ and compute

```
In[4939]:= Do[
  Do[
   If[RSteps + 2 ≤  MaxNumberOfSteps,
     Bound[G, {2}, 2 m, s] = Sum[z[s]^j nrSAW[j, d, {2}], {j, 2 m, Explicit}] +
        d/(d - 1) (2 d z[s])^RSteps VarGamma2[s] Ivalue[1, RSteps + 2, {0}];,
     Bound[G, {2}, 2 m, s] = Sum[z[s]^j nrSAW[j, d, {2}], {j, 2 m, MaxNumberOfSteps - 4}] +
        d/(d - 1) (2 d z[s])^(MaxNumberOfSteps-2) VarGamma2[s] Ivalue[1, MaxNumberOfSteps, {0}];];
   , {m, 1, ComputedSteps / 2}]
  , {s, {i, o}}]
```

We are also interested in bounds on $G^{e_1}_{m,z}(e_1 + e_2)$, that is, the two-point function on the lattice $Z^d \setminus \{e_1\}$. We obtain this bound by removing some of the explicit paths using the vertex e_1.

```
In[4940]:= Do[
  Bound[G, ikNotUsingi, 6, s] = Bound[G, {2}, 6, s] - z[s]^6 36 (d - 2)^2;
  Bound[G, ikNotUsingi, 4, s] =
   z[s]^4 (nrSAW[4, d, {2}] - 2 (d - 2)) + Bound[G, ikNotUsingi, 6, s];
  Bound[G, ikNotUsingi, 2, s] = z[s]^2 + Bound[G, ikNotUsingi, 4, s];
  , {s, {i, o}}]
```

Bound on $G_{m,z}(2\, e_1)$

Now, we compute bounds on $G_{m,z}(2\, e_1)$ and $G^{e_1}_{m,z}(2\, e_1)$. We obtain the latter by subtracting a number of explicit contributions from $G_{m,z}(2\, e_1)$.

```
In[4941]:= Do[
  Do[
   Bound[G, {0, 1}, 2 m, s] = Sum[z[s]^j nrSAW[j, d, {0, 1}], {j, 2 m, Explicit}] +
      (2 d z[s])^RSteps VarGamma2[s] Ivalue[1, RSteps, {0}];
   , {m, 1, ComputedSteps / 2}];
  Bound[G, twoiNotusingi, 6, s] = Bound[G, {0, 1}, 6, s] - z[s]^6 36 (d - 2)^2;
  Bound[G, twoiNotusingi, 4, s] = z[s]^4 2 (d - 1) + Bound[G, twoiNotusingi, 6, s];
  Bound[G, twoiNotusingi, 2, s] = Bound[G, twoiNotusingi, 4, s];
  , {s, {i, o}}]
```

Bound on $\sup_x G_{m,p}(x)$

To compute the supremum of the two-point function, we use that $c_n(x)$=nrSAW[n,d,x] for $n \leq 10$ has its maximal value at either $x = e_1$ or at $x = e_1 + e_2$.

In[4942]:=
```
maxpossible = Max[Explicit, MaxNumberOfSteps / 2];
Do[
 Bound[G, max, maxpossible, s] = (2 d z[s])^maxpossible VarGamma2[o] K[1, maxpossible, {1}];
 Do[
  m = maxpossible - 2 t;
  Bound[G, max, m, s] = Max[nrSAW[m, d, {2}] z[s]^m, nrSAW[m + 1, d, {1}] z[s]^(m+1)] +
    Bound[G, max, m + 2, s];
  , {t, 1, maxpossible / 2 - 1}];
 Bound[G, max, 1, s] = Max[Bound[G, max, 2, s], Bound[G, {1}, 1, s]];
 , {s, {i, o}}]
Clear[t, m, explicit, maxpossible]
```

Bound on $\frac{\overline{\mu}_z}{\mu_z}$

For the general analysis we require an upper on $\frac{\overline{\mu}_z}{\mu_z} = \frac{1}{P(e_i \text{ not in } C^{(0,e_k)}(0))} = \frac{1}{1-\tau_{3,z}(e_1)}$. Since we have upper bound on $\tau_{3,z}(e_1)$=Bound[G,{1},3,s] this is straight forward.

In[4945]:=
```
Do[
 mubOverMu[s] = 1 / (1 - Bound[G, {1}, 3, s]);
 , {s, {i, o}}]
```

Repulsive diagrams

Next, we define bounds on the repulsive diagrams as explained in Section 5.3.2 of (I). As can be seen here, the bounds do no depend on the individual length of the pieces m_1, m_2, ... , but only of the sum of the known length $\sum_i m_i$. So we refer to each diagram by the number of minimal steps $\sum_i m_i$ and the number of two-point function without fixed length (that is, $G_{m,z}(x)$ instead of $G_{\underline{m},z}(x)$).

In[4946]:=

```
Do[
 Bound[Loop, 4, s] = 2 d z[s] Bound[G, {1}, 3, s];
 Do[
  Bound[Bubble, m, s] = Sum[(j + 1 - m) nrBAW[j, d, {0}] z[s]^j, {j, m, Explicit}] +
     (RSteps - m) (2 d z[s])^RSteps VarGamma2[s] Ivalue[1, RSteps, {0}] +
     (2 d z[s])^RSteps VarGamma2[s]^2 Ivalue[2, RSteps, {0}];
  , {m, 2, 8}];
 Do[
  Bound[Triangle, m, s] =
    Sum[1/2 (j + 1 - m) (j + 2 - m) nrBAW[j, d, {0}] z[s]^j, {j, m, Explicit}] +

     1/2 (RSteps - m) (RSteps - 1 - m) (2 d z[s])^RSteps VarGamma2[s] Ivalue[1, RSteps, {0}] +
     (RSteps + 1 - m) (2 d z[s])^RSteps VarGamma2[s]^2 Ivalue[2, RSteps, {0}] +
     (2 d z[s])^RSteps VarGamma2[s]^3 Ivalue[3, RSteps , {0}];
  , {m, 2, 5}];
 Do[
  Bound[Square, m, s] =
    Sum[1/6 (j + 1 - m) (j + 2 - m) (j + 3 - m) nrBAW[j, d, {0}] z[s]^j, {j, m, Explicit}] +
     1/6 (RSteps + 1 - m) (RSteps + 2 - m) (RSteps + 3 - m) (2 d z[s])^RSteps VarGamma2[s]
      Ivalue[1, RSteps, {0}] + 1/2 (RSteps - m) (RSteps - 1 - m) (2 d z[s])^RSteps
      VarGamma2[s]^2 Ivalue[2, RSteps, {0}] +
     (RSteps - m) (2 d z[s]) ^RSteps VarGamma2[s]^3 Ivalue[3, RSteps, {0}] +
     (2 d z[s])^RSteps  VarGamma2[s]^4 Ivalue[4, RSteps, {0}];

  , {m, 2, 4}];
 , {s, {i, o}}]
```

Open repulsive diagrams

We can bound open repulsive diagrams, in the same way. We also compute the bound without extracting contributions, as it is aprior not clear which bound is better. For these bounds we use the monotonicity of the SRW-integrals, see Lemma 5.1 of (I), to conclude that $\sup_{x\neq 0} K_{n,l}(x) = K_{n,l}(e_1)$.

```
In[4947]:= Do[
  Do[
    Bound[OpenBubble, m, s] = Min[(2 d z[s])^m VarGamma2[s]^2 K[2, m, {1}],
      Max[Sum[(j + 1 - m) z[s]^j nrBAW[j, d, {2}], {j, m, ComputedSteps}],
        Sum[(j + 1 - m) z[s]^j nrBAW[j, d, {1}], {j, m, ComputedSteps}]]
       + (ComputedSteps + 1 - m) (2 d z[s])^(ComputedSteps+1) K[1, ComputedSteps + 1, {1}] +
       (2 d z[s])^(ComputedSteps+1) VarGamma2[s]^2 K[2, ComputedSteps + 1, {1}]];
    Bound[OpenTriangle, m, s] = Min[(2 d z[s])^m VarGamma2[s]^3 K[3, m, {1}],
      Max[Sum[1/2 (j + 1 - m) (j + 2 - m) z[s]^j nrBAW[j, d, {2}], {j, m, ComputedSteps}],
         Sum[1/2 (j + 1 - m) (j + 2 - m) z[s]^j nrBAW[j, d, {1}], {j, m, ComputedSteps}]]
       1/2 (ComputedSteps + 1 - m) (ComputedSteps + 1 - 1 - m) (2 d z[s])^(ComputedSteps+1)
        VarGamma2[s] K[1, ComputedSteps + 1, {1}] +
       (ComputedSteps + 1 - m) (2 d z[s])^(ComputedSteps+1) VarGamma2[s]^2
        K[2, ComputedSteps + 1, {1}] + (2 d z[s])^(ComputedSteps+1) VarGamma2[s]^3
        K[3, ComputedSteps + 1, {1}]];
    Bound[OpenSquare, m, s] = Min[(2 d z[s])^m VarGamma2[s]^4 K[4, m, {1}],
      Max[Sum[1/6 (j + 1 - m) (j + 2 - m) (j + 3 - m) z[s]^j nrBAW[j, d, {2}],
          {j, m, ComputedSteps}],
         Sum[1/6 (j + 1 - m) (j + 2 - m) (j + 3 - m) z[s]^j nrBAW[j, d, {1}],
          {j, m, ComputedSteps}]]
       1/6 (10 + 1 - m) (10 + 2 - m) (10 + 3 - m) (2 d z[s])^(ComputedSteps+1) VarGamma2[s]
        K[1, ComputedSteps + 1, {1}] + 1/2 (10 - m) (10 - 1 - m) (2 d z[s])^(ComputedSteps+1)
        VarGamma2[s]^2 K[2, ComputedSteps + 1, {1}] +
       (10 - m) (2 d z[s])^(ComputedSteps+1) VarGamma2[s]^3 K[3, ComputedSteps + 1, {1}] +
       (2 d z[s])^(ComputedSteps+1) VarGamma2[s]^4 K[4, ComputedSteps + 1, {1}]];
    , {m, 1, 4}];
  , {s, {i, o}}]
```

Weighted Diagrams

Here we define bounds on the weighted diagrams at the initial point $z = z_I$, as explained in Section 3.3.3 of (I). For $z = z_I$, the bounds are independent of the bootstrap analysis. These bounds have been implemented in the accompanying notebook ***General.nb***. Concerning notation, as for the unweighted diagrams, we refer to the diagrams by their number of two-point functions and the number of fixed steps on the unweighted lines.

```
In[4948]:= Bound[WeightedBubble, 6, i] = (2 d z[i])^6 BoundFThreeInital[d, 1, 6, 1, {{0}}] ;
           Bound[WeightedOpenLine, 0, i] = BoundFThreeInital[d, 0, 0, 1, {{1}, {2}, {0, 1}}] ;

           Do[
            Bound[WeightedOpenBubble, t, i] =
              (2 d z[i])^t BoundFThreeInital[d, 1, t, 1, {{1}, {2}, {0, 1}}];
            , {t, 0, 3}]
```

For $z \in (z_I, z_c)$, we use the bootstrap function f_3 and the constants $c_{n,l,S}$ to obtain the bounds

```
In[4951]:= Bound[WeightedBubble, 6, o] = (2 d z[o])^6 c[1, 6, 0] * Gamma3;
           Bound[WeightedOpenLine, 0, o] = c[0, 0, 1] * Gamma3;
           Do[
            Bound[WeightedOpenBubble, t, o] = (2 d z[o])^t c[1, t, 1] Gamma3;
            , {t, 0, 3}]
```

As explained in Section 5.3.3 of (I), we dramatically improve the bounds on the closed, weighted, repulsive diagrams by extracting explicit contributions and using repulsiveness. The variable *Bound[WeightedBubble,l,s]* is a bound on $H^{1,l}$, see (3.14) of (I).

```
In[4954]:= Do[
            explicit = Min[MaxNumberOfSteps / 2 - 2, ComputedSteps];
            rem = explicit + 1;

            (*The points x for which we have not computed K(x). We bound its value by K(2e1)>
             K(x).*)
            point5Remainder = {{1, 1}, {0, 0, 1}, {0, 0, 0, 0, 1}, {1, 0, 0, 1}, {0, 1, 1},
              {2, 0, 1}, {1, 2}, {3, 1}};
            point4Remainder = {{0, 1}, {1, 0, 1}, {0, 0, 0, 1}, {0, 2}, {2, 1}};
            Bound[WeightedBubble, 5, s] =
             Bound[WeightedBubble, 6, s] +
              z[s]^5 nrSAW[5, d, {1}] *
               (Sum[ nrSAW[r, d, {1}] z[s]^r, {r, 1, explicit}] + (2 d z[s])^rem K[1, rem, {1}]) +
              3 z[s]^5 nrSAW[5, d, {3}] *
               (Sum[ nrSAW[r, d, {1}] z[s]^r, {r, 1, explicit}] + (2 d z[s])^rem K[1, rem, {2}]) +
              5 z[s]^5 nrSAW[5, d, {5}] *
               (Sum[ nrSAW[r, d, {5}] z[s]^r, {r, 1, explicit}] + (2 d z[s])^rem K[1, rem, {2}]) +
              25 z[s]^5
               Sum[nrSAW[5, d, v] *
                 (Sum[ nrSAW[r, d, v] z[s]^r, {r, 1, explicit}] + (2 d z[s])^rem K[1, rem, {0, 1}]),
                {v, point5Remainder}];

            Bound[WeightedBubble, 4, s] =
             Bound[WeightedBubble, 5, s] +
              2 z[s]^4 nrSAW[4, d, {2}] *
               (Sum[ nrSAW[r, d, {2}] z[s]^r, {r, 1, explicit}] + (2 d z[s])^rem K[1, rem, {2}]) +
```

```
  4 z[s]^4 nrSAW[4, d, {4}] *
    (Sum[ nrSAW[r, d, {2}] z[s]^r, {r, 1, explicit}] + (2 d z[s])^rem K[1, rem, {2}]) +
  16
   Sum[z[s]^4 nrSAW[4, d, v] *
     (Sum[ nrSAW[r, d, v] z[s]^r, {r, 1, explicit}] + (2 d z[s])^rem K[1, rem, {0, 1}]),
    {v, point4Remainder}];

Bound[WeightedBubble, 3, s] =
 Bound[WeightedBubble, 4, s] +
  z[s]^3 nrSAW[3, d, {1}] *
    (Sum[ nrSAW[r, d, {1}] z[s]^r, {r, 1, explicit}] + (2 d z[s])^rem K[1, rem, {1}]) +
  5 z[s]^3 nrSAW[3, d, {1, 1}] *
    (Sum[ nrSAW[r, d, {1, 1}] z[s]^r, {r, 1, explicit}] + (2 d z[s])^rem K[1, rem, {0, 1}]) +
  9 z[s]^3 nrSAW[3, d, {0, 0, 1}] *
    (Sum[ nrSAW[r, d, {0, 0, 1}] z[s]^r, {r, 1, explicit}] + (2 d z[s])^rem K[1, rem, {0, 1}]) +
  3 z[s]^3 nrSAW[3, d, {3}] *
    (Sum[ nrSAW[r, d, {3}] z[s]^r, {r, 1, explicit}] + (2 d z[s])^rem K[1, rem, {2}]);

Bound[WeightedBubble, 2, s] =
 Bound[WeightedBubble, 3, s] +
  (2 d * 4 * z[s]^2 Bound[G, twoiNotusingi, 4, s] + 2 * z[s]^4 2 d (2 d - 2) +
    2 * 2 * z[s]^2 2 d (2 d - 2) Bound[G, ikNotUsingi, 4, s]);
Bound[WeightedBubble, 1, s] = Bound[WeightedBubble, 2, s] + 2 d z[s] Bound[G, {1}, 3, s];
Bound[WeightedBubble, 0, s] = Bound[WeightedBubble, 1, s];

(*remove the help variables from the memory*)
Clear[explicit, rem, point4Remainder, point5Remainder];

, {s, {i, o}}]
```

Building blocks

Blocks without weight

In this Section we define the building blocks, as described in Section 5.1 of (II). A detailed definition the blocks can be found the extended version of (II) that can be found on the arxiv and the homepage of the second author, where you can also download this file. We refer to the extented version of the article as (IIext)

We define the bound on P^b given in Appendix B, Table 5 of (IIext) as follows:

In[4955]:=
```
Do[
  Bound[PS, 0, s] = 1 + 1/2 Bound[Bubble, 2, s];
  Bound[PS, 1, s] = Bound[Loop, 4, s] + Bound[Bubble, 3, s];
  Bound[PS, 2, s] = Bound[Triangle, 4, s] + 1/2 Bound[Bubble, 2, s];
  Bound[PE, 0, s] = 1 + 1/2 Bound[Bubble, 2, s];
  Bound[PE, 1, s] = Bound[Bubble, 3, s];
  Bound[PE, 2, s] = Bound[Triangle, 4, s];
  , {s, {i, o}}];
```

We define the bound on $P^{\iota,b}$appearing in Appendix B, Table 7 of (IIext) as follows:

In[4956]:=
```
Do[
  Bound[Piota, 0, s] = 1/(2 d) + Bound[G, {1}, 3, s] (1 + 1/2 Bound[Bubble, 2, s]);
  (*The 1/(2d) is used to compensate that we also sum over iota when we
   multiplied by A^ι in the bounds.*)
  Bound[Piota, 1, s] = Bound[G, {1}, 3, s] (1 + Bound[PS, 1, s]);
  Bound[Piota, 2, s] = Bound[G, {1}, 3, s] Bound[PS, 2, s] + Bound[G, {1}, 3, s] +
    Bound[Bubble, 4, s]/(2 d z[s]) + (Bound[G, {1}, 3, s] + Bound[Bubble, 3, s] / (2 d z[s]))
    1/2 Bound[Bubble, 2, s];
  , {s, {i, o}}];
```

For $A^{a,b}$ and $A^{a,b,*}$ defined in Appendix B, Table 9 of (IIext) we declare the variables:

```
In[4957]:= Do[
    Bound[A, 0, 0, s] = 1/2 Bound[Bubble, 2, s];
    Bound[A, 0, 1, s] = Bound[Bubble, 3, s];
    Bound[A, 0, 2, s] = Bound[Triangle, 4, s];
    Bound[A, 1, 0, s] = Bound[Bubble, 3, s] / (2 d z[s]);
    Bound[A, 1, 1, s] = Bound[Bubble, 3, s] / (2 d z[s]);
    Bound[A, 1, 2, s] = Bound[Triangle, 4, s] / (2 d z[s]);
    Bound[A, 2, 0, s] = Bound[OpenBubble, 1, s];
    Bound[A, 2, 1, s] = Bound[OpenBubble, 2, s];
    Bound[A, 2, 2, s] = Bound[OpenTriangle, 3, s];

    Bound[ANonRep, 0, 0, s] = 1/2 Bound[Bubble, 2, s];
    Bound[ANonRep, 0, 1, s] = Bound[Bubble, 3, s];
    Bound[ANonRep, 0, 2, s] = Bound[Triangle, 4, s];
    Bound[ANonRep, 1, 0, s] = Bound[Bubble, 3, s] / (2 d z[s]);
    Bound[ANonRep, 1, 1, s] = Bound[Bubble, 3, s] / (2 d z[s]);
    Bound[ANonRep, 1, 2, s] = Bound[Triangle, 4, s] / (2 d z[s]);
    Bound[ANonRep, 2, 0, s] = (2 d z[s])^1 VarGamma2[s]^2 K[2, 1, {1}];
    Bound[ANonRep, 2, 1, s] = (2 d z[s])^2 VarGamma2[s]^2 K[2, 2, {1}];
    Bound[ANonRep, 2, 2, s] = (2 d z[s])^3 VarGamma2[s]^2 K[3, 3, {1}];

    , {s, {i, o}}];
```

The diagrams $A^{\iota,a,b}$ and $A^{\iota,a,b,*}$ are given in Appendix B, Table 9 of (IIext), and are defined as follows:

In[4958]:=
```
Do[
  Bound[Aiota, 0, 0, s] = Bound[Bubble, 3, s];
  Bound[Aiota, 0, 1, s] = Bound[Bubble, 3, s];
  Bound[Aiota, 0, 2, s] = Bound[Triangle, 4, s];
  Bound[Aiota, 1, 0, s] = Bound[Bubble, 3, s] / (2 d z[s]);
  Bound[Aiota, 1, 1, s] = Bound[Bubble, 3, s] / (2 d z[s]);
  Bound[Aiota, 1, 2, s] = Bound[Triangle, 3, s] / (2 d z[s]);
  Bound[Aiota, 2, 0, s] = Bound[OpenBubble, 2, s];
  Bound[Aiota, 2, 1, s] = Bound[OpenBubble, 2, s];
  Bound[Aiota, 2, 2, s] = Bound[OpenTriangle, 3, s];

  Bound[AiotaNonRep, 0, 0, s] = Bound[Bubble, 3, s];
  Bound[AiotaNonRep, 0, 1, s] =
   (2 d z[s])^4 VarGamma2[s] Ivalue[1, 4, {0}] + (2 d z[s])^4 VarGamma2[s]^2 Ivalue[2, 4, {0}];
  Bound[AiotaNonRep, 0, 2, s] = (2 d z[s])^4 VarGamma2[s]^3 Ivalue[3, 4, {0}];
  Bound[AiotaNonRep, 1, 0, s] = (2 d z[s])^3 VarGamma2[s] Ivalue[1, 4, {0}] +
    (2 d z[s])^3 VarGamma2[s]^2 Ivalue[2, 4, {0}];
  Bound[AiotaNonRep, 1, 1, s] = (2 d z[s])^3 VarGamma2[s] Ivalue[1, 4, {0}] +
    (2 d z[s])^3 VarGamma2[s]^2 Ivalue[2, 4, {0}];
  Bound[AiotaNonRep, 1, 2, s] = (2 d z[s])^3 VarGamma2[s]^3 Ivalue[3, 4, {0}];
  Bound[AiotaNonRep, 2, 0, s] = (2 d z[s])^2 VarGamma2[s]^2 K[2, 2, {1}];
  Bound[AiotaNonRep, 2, 1, s] = (2 d z[s])^2 VarGamma2[s]^2 K[2, 2, {1}];
  Bound[AiotaNonRep, 2, 2, s] = (2 d z[s])^3 VarGamma2[s]^3 K[3, 3, {1}];
  , {s, {i, o}}];
```

Then, we define $\overline{A}^{\iota,a,b}$ and $\overline{A}^{\iota,a,b,*}$ as in (B.1)-(B.5) of (IIext) as follows:

```
In[4959]:= Do[
  Do[
   Bound[AiotabarNonRep, a, 0, s] = Bound[AiotaNonRep, a, 0, s];
   Bound[AiotabarNonRep, a, 1, s] = Bound[AiotaNonRep, a, 1, s] / (2 d z[s]);
   , {a, {0, 1, 2}}];
  Bound[AiotabarNonRep, 0, 2, s] = (2 d z[s])^2 VarGamma2[s]^2 K[2, 2, {1}];
  Bound[AiotabarNonRep, 1, 2, s] = (2 d z[s]) VarGamma2[s]^2 K[2, 2, {0}];
  Bound[AiotabarNonRep, 2, 2, s] = 2 d z[s] VarGamma2[s]^2 K[2, 1, {0}];

  Do[
   Bound[AiotaBar, a, 0, s] = Bound[Aiota, a, 0, s];
   Bound[AiotaBar, a, 1, s] = Bound[Aiota, a, 1, s] / (2 d z[s]);
   , {a, {0, 1, 2}}];
  Bound[AiotaBar, 0, 2, s] = Bound[OpenBubble, 2, s];
  Bound[AiotaBar, 1, 2, s] = (2 d z[s]) VarGamma2[s]^2 K[2, 2, {0}];
  Bound[AiotaBar, 2, 2, s] = 2 d z[s] VarGamma2[s]^2 K[2, 1, {0}];
  , {s, {i, o}}];
```

Then, we define $B^{(2),\iota,a,b}$ as in Appendix B, Table 10 of (IIext) as follows:

```
In[4960]:= Do[
  Do[
   Bound[B2i, a, 0, s] = 0;
   Bound[B2i, a, 1, s] = 0;
   , {a, {0, 1, 2}}];
  Bound[B2i, 0, 2, s] = Bound[Bubble, 3, s] Bound[Triangle, 4, s] / (2 d z[s]) +
    Bound[Triangle, 4, s] Bound[OpenTriangle, 3, s];
  Bound[B2i, 1, 2, s] = Bound[Bubble, 3, s] / (2 d z[s]) Bound[Triangle, 4, s] / (2 d z[s]) +
    Bound[Triangle, 4, s] Bound[OpenTriangle, 3, s] / (2 d z[s]);
  Bound[B2i, 2, 2, s] = Bound[Bubble, 3, s] / (2 d z[s]) Bound[OpenTriangle, 3, s] +
    Bound[OpenBubble, 2, s] Bound[OpenTriangle, 4, s];
  , {s, {i, o}}];
```

Further, we define $\overline{B}^{(2),\iota,a,b}$ as in Appendix B, Tables 11&12 of (IIext) as follows:

```
In[4961]:= Do[
  Do[
   Bound[Bbar2i, 0, a, s] = 0;
   , {a, {0, 1, 2}}];
  Bound[Bbar2i, 1, 0, s] = 1/2 Bound[Bubble, 2, s] Bound[OpenTriangle, 3, s];
  Bound[Bbar2i, 1, 1, s] = Bound[Bbar2i, 1, 0, s] / (2 d z[s]);
  Bound[Bbar2i, 1, 2, s] = 1/2 Bound[Bubble, 2, s] Bound[G, max, 1, s]^2;
  Bound[Bbar2i, 2, 0, s] = 1/2 Bound[Bubble, 2, s] Bound[OpenTriangle, 4, s] +
    Bound[Bubble, 3, s]/(2 d z[s]) Bound[Triangle, 4, s] +
    Bound[Triangle, 4, s] Bound[OpenTriangle, 3, s];
  Bound[Bbar2i, 2, 1, s] = Bound[Bbar2i, 2, 0, s] / (2 d z[s]);
  Bound[Bbar2i, 2, 2, s] =
   1/2 Bound[Bubble, 2, s] Bound[G, max, 1, s] Bound[OpenBubble, 2, s] +
    Bound[OpenTriangle, 3, s] Bound[Bubble, 3, s] / (2 d z[s]) +
    Bound[G, max, 1, s] Bound[OpenBubble, 1, s] Bound[Triangle, 4, s];
  , {s, {i, o}}];
```

Definition of diagrams with weight

Now we implement the bound in the diagrams $H^{(1),a,b}$, $H^{(2),\iota,a,b}$ and $H^{(3),\iota,a,b}$ defined in (B.6)-(B.8) of (IIext):

```
In[4962]:= Do[
    Bound[H1, 0, 0, s] = Bound[WeightedBubble, 2, s];
    Bound[H1, 1, 0, s] = Bound[WeightedBubble, 2, s] / (2 d z[s]);
    Bound[H1, 0, 1, s] = Bound[WeightedBubble, 2, s] / (2 d z[s]);
    Bound[H1, 1, 1, s] = Bound[WeightedBubble, 2, s] / (2 d z[s])^2;
    Bound[H1, 2, 0, s] = Bound[WeightedOpenBubble, 0, s];
    Bound[H1, 0, 2, s] = Bound[WeightedOpenBubble, 0, s];

    Bound[H1, 2, 1, s] = Bound[WeightedOpenBubble, 1, s] / (2 d z[s]);
    Bound[H1, 1, 2, s] = Bound[WeightedOpenBubble, 1, s] / (2 d z[s]);
    Bound[H1, 2, 2, s] = Bound[WeightedOpenBubble, 0, s];

    Do[Do[
      Bound[H2, a, b, s] = Bound[H1, a, b, s];
      , {a, {0, 1, 2}}], {b, {0, 1, 2}}];

    Bound[H3, 0, 0, s] = Bound[WeightedBubble, 2, s];
    Bound[H3, 1, 0, s] = Bound[WeightedBubble, 2, s] / (2 d z[s]);
    Bound[H3, 0, 1, s] = Bound[WeightedBubble, 2, s] / (2 d z[s]);
    Bound[H3, 1, 1, s] = Bound[WeightedBubble, 3, s] / (2 d z[s])^2;
    Bound[H3, 2, 0, s] = Bound[WeightedOpenBubble, 1, s];
    Bound[H3, 0, 2, s] = Bound[WeightedOpenBubble, 1, s];

    Bound[H3, 1, 2, s] = Bound[WeightedOpenBubble, 2, s] / (2 d z[s]);
    Bound[H3, 2, 1, s] = Bound[WeightedOpenBubble, 2, s] / (2 d z[s]);
    Bound[H3, 2, 2, s] = Bound[WeightedOpenBubble, 1, s];

    Clear[a, b, t];
    , {s, {i, o}}]
```

As explained in Appendix C.4.1 of (IIext) we bound $C^{(1),\iota,\kappa,a,b}$ and $C^{(2),\iota,\kappa,a,b}$ in terms of other diagrams. We implement $C^{(3),\iota,\kappa,a,b}$ and, for *a=2,b=2,* use the bound (C.44)-(C.49) of (II). For other combinations of *a,b* we improve this bound using three properties:

1.) If *a* or *b* are 1, then we can use symmetry to create an extra $\hat{D}(k)$.

2.) We use that the complete square consists of at least four steps to improve the bound further.

3.) For *a=0* and/or *b=0*, we use that $y \neq u + e_k$ and z≠v.

Using these three properites we obtain the bounds:

```
In[4963]:= Do[

  Bound[C3, 2, 2, s] =
   Bound[WeightedOpenLine, 0, s] (2 d z[s])^2 VarGamma2[s]^3 Ivalue[3, 2, {0}]
     Bound[OpenSquare, 3, s] + Bound[WeightedBubble, 1, s] (2 d z[s]) VarGamma2[s]^2
     Max[Ivalue[2, 2, {0}], K[2, 1, {2}], K[2, 1, {0, 1}]] Bound[OpenTriangle, 2, s];
  Bound[C3, 1, 2, s] =
   Bound[WeightedOpenLine, 0, s] (2 d z[s])^2 VarGamma2[s]^3 Ivalue[3, 2, {0}]
     Bound[Square, 4, s] / (2 d z[s]) + Bound[WeightedBubble, 1, s] (2 d z[s]) VarGamma2[s]^2
     Max[Ivalue[2, 2, {0}], K[2, 1, {2}], K[2, 1, {0, 1}]]
     Bound[Triangle, 4, s] / (2 d z[s]);
  Bound[C3, 2, 1, s] =
   Bound[WeightedOpenLine, 0, s] (2 d z[s])^2 VarGamma2[s]^3 Ivalue[3, 2, {0}]
     Bound[Square, 4, s] / (2 d z[s]) + Bound[WeightedBubble, 1, s] Bound[OpenBubble, 2, s] / (2 d z[s])
     Bound[OpenTriangle, 2, s];
  Bound[C3, 1, 1, s] =
   Bound[WeightedOpenLine, 0, s] Bound[OpenTriangle, 3, s] / (2 d z[s])
     Bound[Square, 4, s] / (2 d z[s]) + Bound[WeightedBubble, 1, s] Bound[OpenBubble, 2, s] / (2 d z[s])
     Bound[Triangle, 3, s] / (2 d z[s]);
  Bound[C3, 0, 1, s] =
   Bound[WeightedOpenLine, 0, s] Bound[OpenTriangle, 3, s] / (2 d z[s]) Bound[Square, 4, s] +
    Bound[WeightedBubble, 1, s] Bound[OpenBubble, 2, s] / (2 d z[s]) Bound[Triangle, 3, s];
  Bound[C3, 1, 0, s] =
   Bound[WeightedOpenLine, 0, s] Bound[OpenTriangle, 3, s] / (2 d z[s]) Bound[Square, 4, s] +
    Bound[WeightedBubble, 1, s] Bound[OpenBubble, 2, s] Bound[Triangle, 3, s] / (2 d z[s]);
  Bound[C3, 0, 0, s] =
   Bound[WeightedOpenLine, 0, s] Bound[OpenTriangle, 3, s] Bound[Square, 4, s] +
    Bound[WeightedBubble, 1, s] Bound[OpenBubble, 2, s] Bound[Triangle, 3, s];
  Bound[C3, 0, 2, s] =
   Bound[WeightedOpenLine, 0, s] Bound[OpenTriangle, 3, s] Bound[OpenSquare, 3, s] +
    Bound[WeightedBubble, 1, s] (2 d z[s]) VarGamma2[s]^2
     Max[Ivalue[2, 2, {0}], K[2, 1, {2}], K[2, 1, {0, 1}]] Bound[Triangle, 3, s];
  Bound[C3, 2, 0, s] =
   Bound[WeightedOpenLine, 0, s] Bound[OpenTriangle, 3, s] Bound[OpenSquare, 3, s] +
    Bound[WeightedBubble, 1, s] Bound[OpenBubble, 2, s] Bound[OpenTriangle, 2, s];
  , {s, {i, o}}]
```

Definition of diagrams specific for initial step iota

Next we bound the terms defined in (5.6)-(5.7) of (II) and bounded in Appendix C.4.2 of (IIext). We bound the terms in Figure 18 one diagram at a time and then sum the three created bounds, see (B.53).

```
In[4964]:= Do[
  (*First time, as explained in the text.*)
  Bound[hi, part1, 0, s] = 1/(2 d) Bound[WeightedBubble, 2, s];
  Bound[hi, part1, 1, s] = Bound[WeightedBubble, 2, s] / ((2 d)^2 z[s]);
  Bound[hi, part1, 2, s] = 1/(2 d) Bound[WeightedOpenBubble, 1, s];
  Do[
   (* The bound for the second term, drawn in Figure 19 and stated in (C.51) *)
   Bound[hi, part2, b, s] = Bound[G, {1}, 3, s] Bound[H2, 0, b, s] +
     2 Bound[G, {1}, 3, s]
      (Sum[Bound[PS, a, s] Bound[H2, a, b, s], {a, 0, 2}] - Bound[H2, 0, b, s]) +
     2 Bound[G, {1}, 3, s]
      (1/2 Bound[WeightedBubble, 1, s] Bound[AiotaNonRep, b, 0, s] +
        Bound[WeightedBubble, 1, s] / (2 d z[s]) Bound[AiotaNonRep, b, 1, s] +
        Min[Bound[WeightedOpenBubble, 0, s] Bound[AiotaNonRep, b, 2, s] (*a=2,
         as stated*), Bound[WeightedOpenBubble, 1, s] Bound[AiotaNonRep, b, 2, s]
          (*a=2, u≠e_i, so line is non-trivial*) +
          1/2 Bound[WeightedBubble, 1, s] Bound[AiotabarNonRep, b, 2, s] (*a=2,
         u=e_i*), Bound[WeightedOpenBubble, 1, s] Bound[AiotaNonRep, b, 2, s]
          (*a=2, u≠e_i, so line is non-trivial*) +
          1/2 Bound[WeightedOpenLine, 0, s] Bound[AiotaNonRep, b, 2, s] (*a=2,
         u=e_i, different decompostion*)]);
   (* The bound for the second term, drawn in Figure 20 and stated in (C.52) *)
   Bound[hi, part3, b, s] = Bound[G, {1}, 3, s] Bound[H2, 1, b, s] (*b=e_i,
     (w,y) neighbors *) + Bound[Bubble, 4, s] / (2 d z[s]) Bound[H2, 2, b, s] (*b=e_i,
     (w,y) further away *) + 2 (Bound[G, {1}, 3, s] + Bound[Bubble, 3, s] / (2 d z[s]))
     (1/2 Bound[Bubble, 2, s] Max[Bound[H2, 1, b, s], Bound[H2, 2, b, s]] +
       1/2 Bound[WeightedBubble, 1, s] Max[Bound[AiotabarNonRep, 1, b, s],
         Bound[AiotabarNonRep, 2, b, s]]);
   (* This is (B.46) *)
   Bound[hi, b, s] = Sum[Bound[hi, t, b, s], {t, {part1, part2, part3}}]
   , {b, 0, 2}];
  , {s, {i, o}}]
```

We define the bound for $h^{i,\mathrm{II},b}$ defined in (5.7) and bounded in (C.50) of (IIext).

In[4965]:=
```
Do[
 Do[
   Bound[hII, b, s] =
    Bound[hi, b, s] +
     2 Sum[Bound[hi, a, s] Bound[AiotaNonRep, a, b, s] +
         Sum[Bound[Piota, a, s] Bound[AiotaNonRep, a, c, s] Bound[H2, c, b, s], {c, 0, 2}],
        {a, 0, 2}]
   , {b, 0, 2}];
 , {s, {i, o}}]
```

Definition of vectors and matrices

We define the matrices we use to compute the bounds, see the end of Section 5.1. First, we define the matrices whose entries we have already implemented.

```
In[4966]:= Do[
    Vector[PS, s] = Table[Bound[PS, a, s], {a, {0, 1, 2}}];
    Vector[PE, s] = Table[Bound[PE, a, s], {a, {0, 1, 2}}];
    Vector[PSNT, s] = Vector[PS, s] - {1, 0, 0};
    Vector[PENT, s] = Vector[PE, s] - {1, 0, 0};
    Vector[Piota, s] = Table[Bound[Piota, a, s], {a, {0, 1, 2}}];

    Vector[hS, s] = Table[Bound[H1, 0, b, s], {b, {0, 1, 2}}];
    Vector[hE, s] = Table[Bound[H3, 0, b, s], {b, {0, 1, 2}}];
    Vector[hi, s] = Table[Bound[hi, b, s], {b, {0, 1, 2}}];
    Vector[hII, s] = Table[Bound[hII, b, s], {b, {0, 1, 2}}];



    Matrix[A, s] = Table[Bound[A, a, b, s], {a, {0, 1, 2}}, {b, {0, 1, 2}}];
    Matrix[ANonRep, s] = Table[Bound[A, a, b, s], {a, {0, 1, 2}}, {b, {0, 1, 2}}];

    Matrix[Aiota, s] = Table[Bound[Aiota, a, b, s], {a, {0, 1, 2}}, {b, {0, 1, 2}}];
    Matrix[AiotaNonRep, s] = Table[Bound[AiotaNonRep, a, b, s], {a, {0, 1, 2}},
      {b, {0, 1, 2}}];

    Matrix[AiotaBar, s] = Table[Bound[AiotaBar, a, b, s], {a, {0, 1, 2}}, {b, {0, 1, 2}}];
    Matrix[AiotabarNonRep, s] = Table[Bound[AiotabarNonRep, a, b, s], {a, {0, 1, 2}},
      {b, {0, 1, 2}}];

    Matrix[B2i, s] = Table[Bound[B2i, a, b, s], {a, {0, 1, 2}}, {b, {0, 1, 2}}];
    Matrix[Bbar2i, s] = Table[Bound[Bbar2i, a, b, s], {a, {0, 1, 2}}, {b, {0, 1, 2}}];

    Matrix[C3, s] = Table[Bound[C3, a, b, s], {a, {0, 1, 2}}, {b, {0, 1, 2}}];
    Matrix[H1, s] = Table[Bound[H1, a, b, s], {a, {0, 1, 2}}, {b, {0, 1, 2}}];
    Matrix[H2, s] = Table[Bound[H2, a, b, s], {a, {0, 1, 2}}, {b, {0, 1, 2}}];
    Matrix[H3, s] = Table[Bound[H3, a, b, s], {a, {0, 1, 2}}, {b, {0, 1, 2}}];
    , {s, {i, o}}]
```

Next, we define the bound for B and $\overline{B}$, see (5.4)-(5.5) of (II):

```
In[4967]:= Do[
    Matrix[B, s] = Matrix[AiotaNonRep, s].Matrix[A, s] + Matrix[Aiota, s] * Bound[PS, 0, s] +
      Matrix[B2i, s];
    Matrix[Bbar, s] = Matrix[Aiota, s].Matrix[ANonRep, s] + Matrix[Aiota, s] +
      Matrix[Bbar2i, s];
    , {s, {i, o}}]
```

Further, we define the bound on $C^{(1)}$ and $C^{(2)}$ as stated in Appendix C.4.1 of (II):

In[4968]:=
```
Do[
 Matrix[C1, s] =
  (2 Matrix[H2, s].Matrix[A, s] + 2 Matrix[AiotaNonRep, s].Matrix[H1, s] + Matrix[H2, s]).
    Matrix[Aiota, s] +
   Matrix[H2, s].Matrix[Bbar2i, s];
 Matrix[C2, s] =
  (2 Matrix[H3, s].Matrix[A, s] + 2 Matrix[AiotaNonRep, s].Matrix[H1, s] + Matrix[H3, s]).
    Matrix[Aiota, s] + 2 Matrix[C3, s] + Matrix[H3, s].Matrix[Bbar2i, s];
 , {s, {i, o}}]
```

We compute the eigensystem of the matices B and $\overline{B}$ to sum the matrix-valued bounds, as described in Section 5.4 of (I).

In[4969]:=
```
Do[
 EigensystemB[s] = Eigensystem[Transpose[Matrix[B, s]]];
 EigensystemBbar[s] = Eigensystem[Matrix[Bbar, s]];
 InverseProductB[s] = Inverse[Transpose[EigensystemB[s][[2]]]].Vector[PS, s];
 InverseProductBForPiota[s] =
  Inverse[Transpose[EigensystemB[s][[2]]]].Vector[Piota, s];
 InverseProductBbar[s] = Inverse[Transpose[EigensystemBbar[s][[2]]]].Vector[PE, s];

 Do[
  EigenVectorB[j, s] = EigensystemB[s][[2, j]] * InverseProductB[s][[j]];
  EigenVectorBbar[j, s] = EigensystemBbar[s][[2, j]] * InverseProductBbar[s][[j]];
  EigenVectorBForPiota[j, s] =
   EigensystemB[s][[2, j]] * InverseProductBForPiota[s][[j]];

  EigenValueB[j, s] = EigensystemB[s][[1, j]];
  EigenValueBbar[j, s] = EigensystemBbar[s][[1, j]];
  EigenValueBForPiota[j, s] = EigensystemB[s][[1, j]];
  , {j, 1, 3}];
 , {s, {i, o}}]
```

Bound on the coefficients

Here, we implement the bounds on the coefficients of Assumption 4.3 of (I) that are stated in Sections 4.3, 5.2, 5.3 of (II).

Bounds for $N=0$

The bound stated in Lemma 4.2 of (II) is as follows:

```
In[4970]:= Do[
    Bound[Xi, 0, s] = 1/2 Bound[Bubble, 2, s];
    Bound[Xi, 0, Delta, s] = 1/2 Bound[WeightedBubble, 1, s];
    Bound[Xi, R, 0, s] = 1/2 Bound[Bubble, 4, s];
    Bound[Xi, R, 0, Delta, s] = 1/2 Bound[WeightedBubble, 2, s];

    Bound[Psi, RII, 0, s] = 1/2 Min[1, mubOverMu[s] (d - 1)/d] Bound[Bubble, 4, s];
    Bound[Psi, RII, 0, Delta, s] =
     1/2 Min[1, mubOverMu[s] (d - 1)/d] Bound[WeightedBubble, 2, s];

    Bound[Psi, RI, 0, s] = (2 d - 2) z[s] Bound[G, {1}, 3, s] +
      1/2 Min[1, mubOverMu[s] (d - 1)/d] Bound[Bubble, 4, s];
    Bound[Psi, RI, 0, Delta, s] =
     mubOverMu[s]/2 (Bound[Bubble, 2, s] + Bound[WeightedBubble, 1, s]);
    , {s, {i, o}}]
```

Next, we implement the bounds stated in Lemma 4.3 of (II):

```
In[4971]:= Do[
  Bound[XiIota, 0, s] = Bound[G, {1}, 3, s] (1 + 1/2 Bound[Bubble, 2, s]);
  Bound[XiIota, 0, Delta, ei, s] = 1/2 Bound[G, {1}, 3, s] Bound[WeightedBubble, 1, s];
  Bound[XiIota, 0, Delta, 0, s] =
   Bound[G, {1}, 3, s] (1 + 1/2 Bound[Bubble, 2, s] + 1/2 Bound[WeightedBubble, 1, s]);

  Bound[XiIota, alphaI, 0, Atei, s] = Bound[G, {1}, 3, s];
  Bound[XiIota, alphaII, 0, AtZero, s] = 0;

  Bound[XiIota, alphaI, 0, SumAroundei, s] = Bound[G, {1}, 3, s]^2 (2 d - 1) z[s];
  Bound[XiIota, alphaII, 0, SumAroundZero, s] = Bound[G, {1}, 3, s];

  Bound[XiIota, RI, 0, s] = 1/2 Bound[G, {1}, 3, s] Bound[Bubble, 4, s];
  Bound[XiIota, RI, 0, Delta, ei, s] =
   1/2 Bound[G, {1}, 3, s] Bound[WeightedBubble, 2, s];
  Bound[XiIota, RII, 0, s] = 1/2 Bound[G, {1}, 3, s] Bound[Bubble, 2, s];
  Bound[XiIota, RII, 0, Delta, 0, s] =
   Bound[G, {1}, 3, s]/2 (Bound[Bubble, 2, s] + Bound[WeightedBubble, 1, s]);
  Bound[Pi, alpha, 0, s] = (2 d - 2) z[s] Bound[G, {1}, 3, s];
  Bound[Pi, R, 0, s] = 2 d^2 z[s] Bound[G, {1}, 3, s] Bound[Bubble, 2, s];
  Bound[Pi, R, 0, Delta, eiek, s] =
   2 d^2 z[s] Bound[G, {1}, 3, s] (Bound[Bubble, 2, s] + Bound[WeightedBubble, 1, s]);
  , {s, {i, o}}]
```

Bounds for *N=1*

Implementation of the bounds stated in Lemma 5.1 and 5.2 of (II):

```
In[4972]:= Do[
   (* These are the bounds in Lemma 5.1 *)
   Bound[Xi, 1, s] = Vector[PS, s].Matrix[AiotaBar, s].Vector[PE, s];
   Bound[Xi, R, 1, s] = Vector[PS, s].Matrix[AiotaBar, s].Vector[PE, s] -
     Bound[AiotaBar, 0, 0, s] + Bound[Bubble, 4, s];
   Bound[Psi, RI, 1, s] =
    (2 d - 1)/(2 d) mubOverMu[s]
     (Vector[PS, s].Matrix[AiotaBar, s].Vector[PE, s] - Bound[AiotaBar, 0, 0, s] +
       (2 d - 2) z[s] Bound[G, {1}, 3, s] + Bound[Bubble, 4, s]);
   Bound[Psi, RII, 1, s] =
```

```
 2 d - 1
 ------- mubOverMu[s]
   2 d
  (Vector[PS, s].Matrix[AiotaBar, s].Vector[PE, s] - Bound[AiotaBar, 0, 0, s] +
    Bound[Bubble, 4, s]);

betatmp = Bound[H3, 0, 0, s] +
   2 * (1/2 Bound[Bubble, 2, s] Bound[WeightedBubble, 2, s] +
      1/2 Bound[WeightedBubble, 1, s] Bound[Bubble, 3, s])
     (*Case a) u=w (Reference to Appendix B.1 of (II)*) +
   1/2 Bound[Bubble, 2, s] Bound[WeightedOpenBubble, 2, s](*Case b) w=0*) +
   4 Bound[WeightedBubble, 3, s] / (2 d z[s]) Bound[Bubble, 3, s] +
4 Bound[WeightedOpenBubble, 2, s] / (2 d z[s]) Bound[Bubble, 4, s]
   (*Case c) 2d D(u,w)=1*) +
   4 Bound[WeightedOpenBubble, 2, s] Bound[Triangle, 4, s] +
4 Bound[WeightedOpenBubble, 1, s] Bound[Triangle, 5, s](*Case d) 2d D(u,w)=1*) +
   3 Vector[hS, s].Transpose[Matrix[Aiota, s]].Vector[PENT, s] +
   3 Vector[PSNT, s].Matrix[H3, s].Vector[PENT, s] +
   3 Vector[PSNT, s].Matrix[Aiota, s].Vector[hE, s];


Bound[Xi, 1, Delta, s] = betatmp;
Bound[Xi, R, 1, Delta, s] =
 betatmp - Bound[H2, 0, 0, s] + Bound[WeightedBubble, 2, s];
Bound[Psi, RI, 1, Delta, s] =
 mubOverMu[s] (betatmp + Vector[PS, s].Matrix[AiotaBar, s].Vector[PE, s]);
Bound[Psi, RII, 1, Delta, s] =
 mubOverMu[s] (betatmp - Bound[H2, 0, 0, s] + Bound[WeightedBubble, 2, s]);

(* These are the bounds in Lemma 5.2 *)
Bound[XiIota, 1, s] = Vector[Piota, s].Matrix[AiotaBar, s].Vector[PE, s];
Bound[XiIota, 1, Delta, ei, s] =
 Bound[hi, 0, s] + 2 Vector[hi, s].Vector[PENT, s] +
  2 Vector[Piota, s].Matrix[Aiota, s].Vector[hE, s] -
  (Vector[hi, s])[[1]] (Vector[PENT, s])[[1]] -
  (Vector[Piota, s].Matrix[Aiota, s])[[1]] (Vector[hE, s])[[1]];
Bound[XiIota, 1, Delta, 0, s] =
 2 Bound[XiIota, 1, Delta, ei, s] + 2 Bound[XiIota, 1, s];

Clear[betatmp];
, {s, {i, o}}];
```

Lower bounds

Next we incorporate the bounds stated in Lemma 5.3 of (II):

```
In[4973]:= Do[
  theta2 = Max[Bound[G, ikNotUsingi, 2, s], Bound[G, twoiNotusingi, 2, s]];
  theta4 = Max[Bound[G, ikNotUsingi, 4, s], Bound[G, twoiNotusingi, 4, s]];
  vartheta = d^2/((d - 1) (d - 2)) (2 d z[s])^5 VarGamma2[s] Ivalue[1, 8, {0}];
  Bound[Pi, alpha, lower, 0, s] =
   (2 d - 1) (2 d - 2) z[i]^4 (1 - z[i]^3)^(2 d-3) - (2 d - 2)^2 z[s]^4 Bound[G, {1}, 3, s]^2 -
    (2 d - 2) z[s]^4 (Bound[G, twoiNotusingi, 2, s] + (4 d - 5) Bound[G, ikNotUsingi, 2, s] +
       (4 d - 4) z[s]^3 + 2 d vartheta) +
    16 (d - 1) (d - 2) (2 d - 3) z[i]^6 (1 - z[i]^3)^(2 d-2) (1 - z[i]^5)^(16 (d-1) (d-2)-1)
     (1 - Bound[G, {1}, 3, s] - Bound[G, ikNotUsingi, 2, s] - 3 Bound[G, max, 1, s] );

  Bound[Psi, lower, 0, s] = (2 d - 1) (2 d - 2) z[i]^4 (1 - z[i]^3)^(2 d-3) -
    (2 d - 2)^2 z[s]^4 Bound[G, {1}, 3, s]^2 -
    (2 d - 2) z[s]^4 (Bound[G, twoiNotusingi, 2, s] + (4 d - 5) Bound[G, ikNotUsingi, 2, s] +
       (4 d - 4) z[s]^3 + 2 d vartheta) +
    (2 d - 2)^2 z[i]^4 (1 - Bound[G, {1}, 3, s] - 2 theta2 - 2 z[s]) - 2 d (2 d - 2) z[i]^4 vartheta +
    4 * 16 d (d - 1) (d - 2) z[i]^6 (1 - z[i]) (1 - z[i]^3)^(2 d-2) (1 - z[i]^5)^(16 (d-1) (d-2))
     (1 - 2 z[s]^2 - Bound[G, {1}, 3, s] - 2 theta2 - 2 vartheta -
       d/(d - 1) (2 d z[s])^5 VarGamma2[s] K[1, 6, {0, 1}]);
  Bound[Pi, 1, Lower, s] =
   (2 d - 1) (2 d - 2) z[i]^5 (1 - z[s] - 3 Bound[G, {1}, 3, s] - theta2 - theta4) -
    (2 d - 2) z[i]^5 (theta4 + vartheta) (* (B.3) *) +
    (2 d - 2)^2 (2 d - 3) z[i]^7 (1 - z[i]^3)^(2 d-3)
     (1 - z[s] - 2 z[s]^2 - 2 z[s]^3 - 2 Bound[G, {1}, 3, s] - 4 theta4 - 2 vartheta)
     (1 - Bound[G, {1}, 3, s] - theta2 - vartheta) (* (B.10) *) +
    (2 d - 2)^2 (2 d - 3) z[i]^7 (1 - z[i]^3)^(2 d-3)
     (1 - z[s] - z[s]^2 - 2 Bound[G, {1}, 3, s] - 2 theta4 - vartheta)
     (1 - Bound[G, {1}, 3, s] - theta2) (* (B.12) *) +
    (2 d - 2)^3 (2 d - 3) z[i]^8 (1 - z[i]^3)^(2 d-3)
     (1 - z[s] - 2 z[s]^2 - 2 Bound[G, {1}, 3, s] - 4 theta4 - 3 vartheta -
       (2 d z[s])^4 VarGamma2[s] Ivalue[1, 8, {0}])
     (1 - 2 Bound[G, {1}, 3, s] - theta2 - vartheta) (* (B.15) *);
  Clear[theta2, theta4, vartheta]
  , {s, {i, o}}];
```

Bounds for differences

We continue with the bounds stated in Lemma 5.4 of (II):

```
In[4974]:= Do[
  theta2 = Max[Bound[G, ikNotUsingi, 2, s], Bound[G, twoiNotusingi, 2, s]];
```

```
theta4 = Max[Bound[G, ikNotUsingi, 4, s], Bound[G, twoiNotusingi, 4, s]];
vartheta = d^2/((d - 1) (d - 2)) (2 d z[s])^5 VarGamma2[s] Ivalue[1, 8, {0}];
Bound[Xi, alpha, OneMinusZero, AtZero, s] = 0;
Bound[Xi, alpha, ZeroMinusOne, AtZero, s] = 0;

Bound[Xi, alpha, OneMinusZero, AtEi, s] =
  (2 d - 2) z[s]^4 (1 - (1 - z[s]^3)^(2 d-3)) +
   z[s]^2 ((2 d - 2) Bound[G, ikNotUsingi, 4, s] + Bound[G, twoiNotusingi, 4, s]);
Bound[Xi, alpha, ZeroMinusOne, AtEi, s] = z[s] Bound[G, {1}, 5, s] + (2 d - 2) z[s]^5 +
   4 (d - 1) z[s]^4 (Bound[G, {1}, 3, s] + Bound[G, ikNotUsingi, 4, s]);

Bound[Psi, alphaI, OneMinusZero, AroundEi, s] =
 mubOverMu[s] z[s]^2
    ((2 d - 2) z[s]^2 + 4 (d - 1) Bound[G, ikNotUsingi, 4, s] +
      Bound[G, twoiNotusingi, 2, s]) +
   (2 d - 2) mubOverMu[s] z[s]^4
    (Bound[G, {1}, 3, s] + Bound[G, ikNotUsingi, 2, s] + Bound[G, twoiNotusingi, 2, s] +
      d/(d - 1) (2 d z[s])^5 VarGamma2[s] K[1, 6, {0, 1}]);

Bound[Psi, alphaI, ZeroMinusOne, AroundEi, s] =
 mubOverMu[s] z[s]^2
    ((2 d - 2) z[s]^2 + 4 (d - 1) Bound[G, ikNotUsingi, 4, s] + Bound[G, twoiNotusingi, 2, s] +
      (4 d - 3) Bound[G, {1}, 3, s]^2) +
   (2 d - 2) mubOverMu[i] z[i]^4
    (1 - 2 (1 - 2 Bound[G, {1}, 3, s] - Bound[G, ikNotUsingi, 2, s])
       (1 - Bound[G, {1}, 3, o] - theta2));


Bound[Psi, alphaII, OneMinusZero, AroundZero, s] =
  (2 d - 1) (2 d - 2) mubOverMu[s] z[s]^4 (1 - (1 - z[s]^3)^(2 d-3)) +
   (2 d - 1) mubOverMu[s] z[s]^2
    ((2 d - 2) Bound[G, ikNotUsingi, 4, s] + Bound[G, twoiNotusingi, 4, s]) +
   (2 d - 2)^2 z[s]^4 Bound[G, {1}, 3, s]^2 +
   (2 d - 2) z[s]^4 (Bound[G, twoiNotusingi, 4, s] + (4 d - 5) Bound[G, ikNotUsingi, 2, s] +
      (4 d - 4) z[s]^3 + 2 d vartheta);

Bound[Psi, alphaII, ZeroMinusOne, AroundZero, s] =
  (2 d - 1) z[s] Bound[G, {1}, 5, s] +
   (2 d - 1) (2 d - 2) z[s]^4 mubOverMu[s]
    (1 - (1 - z[s] - 2 Bound[G, {1}, 3, o] - 2 theta4)
       (1 - Bound[G, {1}, 3, o] - theta2 - vartheta));
```

```
Clear[theta2, theta4, vartheta]
, {s, {i, o}}];
```

Bounds for *N=2,3*

We consider also *N=2,3* as a special case as we improve the bound on the weighted diagrams stated in Proposition 5.5 and 5.6 of (II) slightly. We extract the cases in which the inital and last triangle are trivial. In this case we can reduce the leading factors *(N+1)* and *(N+2)*, respectively.

```
In[4975]:= Do[
  Bound[Xi, 2, s] = Vector[PS, s].Matrix[B, s].Matrix[AiotaBar, s].Vector[PE, s];
  Bound[XiIota, 2, s] = Vector[Piota, s].Matrix[B, s].Matrix[AiotaBar, s].
    Vector[PE, s];
  Bound[Xi, 3, s] = Vector[PS, s].Matrix[B, s].Matrix[B, s].Matrix[AiotaBar, s].
    Vector[PE, s];
  Bound[XiIota, 3, s] = Vector[Piota, s].Matrix[B, s].Matrix[B, s].
    Matrix[AiotaBar, s].Vector[PE, s];

  Bound[Xi, 3, Delta, s] =
   3 ({1, 0, 0}.Matrix[B, s].Matrix[B, s].Matrix[H3, s].{1, 0, 0} +
       {1, 0, 0}.(Matrix[B, s].Matrix[C1, s] + Matrix[C2, s].Matrix[Bbar, s]).{1, 0, 0}) +
    4
     ({1, 0, 0}.Matrix[B, s].Matrix[B, s].
        (Matrix[H2, s].Vector[PENT, s] + Matrix[Aiota, s].Vector[hE, s]) +
       {1, 0, 0}.(Matrix[C1, s].Matrix[Bbar, s] + Matrix[B, s].Matrix[C2, s]).
        Vector[PENT, s]) +
    4 (Vector[hS, s].Transpose[Matrix[Aiota, s]].Matrix[Bbar, s].Matrix[Bbar, s].
        {1, 0, 0} + Vector[PSNT, s].Matrix[B, s].Matrix[B, s].(Matrix[H2, s].{1, 0, 0}) +
       Vector[PSNT, s].(Matrix[C1, s].Matrix[Bbar, s] + Matrix[B, s].Matrix[C2, s]).
        {1, 0, 0}) +
    5 (Vector[hS, s].Transpose[Matrix[Aiota, s]].Matrix[Bbar, s].Matrix[Bbar, s].
        Vector[PENT, s] + Vector[PSNT, s].Matrix[B, s].Matrix[B, s].
        (Matrix[H2, s].Vector[PENT, s] + Matrix[Aiota, s].Vector[hE, s]) +
       Vector[PSNT, s].(Matrix[C1, s].Matrix[Bbar, s] + Matrix[B, s].Matrix[C2, s]).
        Vector[PENT, s]);

  Bound[XiIota, 2, Delta, ei, s] =
   2 (Vector[hII, s].Matrix[Bbar, s].{1, 0, 0} +
       Vector[Piota, s].Matrix[B, s].Matrix[H3, s].{1, 0, 0}) +
    3 (Vector[hII, s].Matrix[Bbar, s].Vector[PENT, s] +
       Vector[Piota, s].Matrix[B, s].
        (Matrix[H3, s].Vector[PENT, s] + Matrix[Aiota, s].Vector[hE, s]));

  Bound[XiIota, 3, Delta, ei, s] =
   3 (Vector[hII, s].Matrix[Bbar, s].Matrix[Bbar, s].{1, 0, 0} +
```

```
      Vector[Piota, s].Matrix[B, s].Matrix[B, s].Matrix[H2, s].{1, 0, 0}) +
   3 Vector[Piota, s].Matrix[B, s].Matrix[C2, s].{1, 0, 0} +
   4 (Vector[hII, s].Matrix[Bbar, s].Matrix[Bbar, s].Vector[PENT, s] +
      Vector[Piota, s].Matrix[B, s].Matrix[B, s].
       (Matrix[H2, s].Vector[PENT, s] + Matrix[Aiota, s].Vector[hE, s])) +
   4 Vector[Piota, s].Matrix[B, s].Matrix[C2, s].Vector[PENT, s];


Bound[XiIota, 2, Delta, 0, s] =
  3 Vector[Piota, s].Matrix[B, s].Matrix[AiotaBar, s].{1, 0, 0} +
   4 Vector[Piota, s].Matrix[B, s].Matrix[AiotaBar, s].Vector[PENT, s] +
   3 (Vector[hII, s].Matrix[Bbar, s].{1, 0, 0} +
      Vector[Piota, s].Matrix[B, s].Matrix[H3, s].{1, 0, 0}) +
   4 (Vector[hII, s].Matrix[Bbar, s].Vector[PENT, s] +
      Vector[Piota, s].Matrix[B, s].
       (Matrix[H3, s].Vector[PENT, s] + Transpose[Matrix[Aiota, s]].Vector[hE, s]));

Bound[XiIota, 3, Delta, 0, s] =
  4 Vector[Piota, s].Matrix[B, s].Matrix[B, s].Matrix[AiotaBar, s].{1, 0, 0} +
   5 Vector[Piota, s].Matrix[B, s].Matrix[B, s].Matrix[AiotaBar, s].Vector[PENT, s] +
   4 (Vector[hII, s].Matrix[Bbar, s].Matrix[Bbar, s].{1, 0, 0} +
      Vector[Piota, s].Matrix[B, s].Matrix[B, s].Matrix[H2, s].{1, 0, 0}) +
   4 Vector[Piota, s].Matrix[B, s].Matrix[C2, s].{1, 0, 0} +
   5 (Vector[hII, s].Matrix[Bbar, s].Matrix[Bbar, s].Vector[PENT, s] +
      Vector[Piota, s].Matrix[B, s].Matrix[B, s].
       (Matrix[H2, s].Vector[PENT, s] + Matrix[Aiota, s].Vector[hE, s])) +
   5 Vector[Piota, s].Matrix[B, s].Matrix[C2, s].Vector[PENT, s];
, {s, {i, o}}]
```

Omit above is the bound on the weighted diagram of $\Xi^{(2)}$ we improve the bound on the weighted diagram once more. We improve this bound as explained in Appendix C.1 of (II). We first bound the weighted diagram, H' (x), defined in (C.6) and bounded in the Remark at the end of Appendix C.1.

```
In[4976]:= Do[
  (* Firt the implicit solution of (B.19)+(B.20)*)
  Bound[Hdash, AtZero, 1, s] =
   Bound[WeightedBubble, 1, s] / (1 - Bound[OpenBubble, 1, s]) +
    (Bound[WeightedOpenBubble, 1, s] Bound[Bubble, 2, s]) / (1 - Bound[OpenBubble, 1, s]);
  (* Firt the implicit solution of (B.19)+(B.20) for over values of x and l*)
  Bound[Hdash, AtZero, 2, s] =
   Bound[WeightedBubble, 2, s] + Bound[Hdash, AtZero, 1, s] Bound[OpenBubble, 2, s] +
    Bound[Bubble, 2, s] Bound[WeightedOpenBubble, 2, s];
  Do[
   Bound[Hdash, NotZero, l, s] = Bound[WeightedOpenBubble, l, s] +
      Bound[Hdash, AtZero, 1, s] Bound[OpenBubble, l, s] +
      Bound[Bubble, 2, s] Bound[WeightedOpenBubble, l, s];
   , {l, 1, 2}]
  , {s, {i, o}}]

In[4977]:= Do[
  Bound[Xi, 2, Delta, s] =
    2 ({1, 0, 0}.(Matrix[B, s] - Matrix[Aiota, s]).Matrix[H3, s].{1, 0, 0} +
        {1, 0, 0}.(Matrix[C1, s] - Matrix[H2, s].Matrix[Aiota, s]).{1, 0, 0}) +
     (*Diagram with left and right triangle trivial minus minus the special
      case drawn in Figure 15.*)
     (Bound[Bubble, 3, s] Bound[Hdash, NotZero, 1, s](*(B.7)*) +
       Bound[Bubble, 3, s] / (2 d z[s]) Bound[Hdash, NotZero, 2, s](*(B.8)*) +
       2 Bound[Bubble, 3, s] Bound[WeightedBubble, 2, s](*(B.9)*) +
       2 (Bound[WeightedBubble, 3, s] / (2 d z[s])) Bound[Bubble, 4, s](*B.10,
       part involving H2*) + 2 (Bound[Hdash, AtZero, 2, s] / (2 d z[s])) Bound[Bubble, 4, s]
       (*B.10, H{*) + 2 Bound[WeightedOpenBubble, 2, s] Bound[Triangle, 5, s]
       (*B.11, H2 part*) + 2 Bound[Hdash, NotZero, 1, s] Bound[Triangle, 5, s]
       (*B.11, H{ *)) +
     6 (Vector[hS, s].Transpose[Matrix[Aiota, s]].Matrix[Bbar, s].{1, 0, 0} +
        Vector[PSNT, s].Matrix[B, s].Matrix[H3, s].{1, 0, 0} +
        Vector[PSNT, s].Matrix[C1, s].{1, 0, 0})
       (*Diagram in which either the left or the right diagram a trivial*) +
     4 (Vector[hS, s].Transpose[Matrix[Aiota, s]].Matrix[Bbar, s].Vector[PENT, s] +
        Vector[PSNT, s].Matrix[B, s].
         (Matrix[H3, s].Vector[PENT, s] + Matrix[Aiota, s].Vector[hE, s]) +
        Vector[PSNT, s].Matrix[C1, s].Vector[PENT, s])
       (*Bound for the diagram in which the left AND the right traingle are non-
    trivial*);
  , {s, {i, o}}]
```

Bounds for $N \geq 4$

Next, we bound the sum over all odd and even $N \geq 4$. We use the earlier computed decomposition of the vectors P^S and P^i in term of eigenvectors of B. We use these eigenvectors and the geometric sum to compute the sum of the bounds, see Section 5.4 of (I). We begin with the bound on the absolute value:

```
In[4978]:= Do[
  Do[
   v[j] = EigenVectorB[j, s];
   vi[j] = EigenVectorBForPiota[j, s];
   e[j] = EigenValueB[j, s];
   evi[j] = EigenValueBForPiota[j, s];
   , {j, {1, 2, 3}}];
  Bound[Xi, EvenTail, s] =
   Bound[Xi, 2, s] + Sum[e[j]^3/(1 - e[j]^2) v[j].Matrix[AiotabarNonRep, s].Vector[PE, s],
     {j, {1, 2, 3}}];
  Bound[Xi, OddTail, s] =
   Bound[Xi, 3, s] + Sum[e[j]^4/(1 - e[j]^2) v[j].Matrix[AiotabarNonRep, s].Vector[PE, s],
     {j, {1, 2, 3}}];
  Bound[XiIota, EvenTail, s] =
   Bound[XiIota, 2, s] + Sum[evi[j]^3/(1 - evi[j]^2) vi[j].Matrix[AiotabarNonRep, s].Vector[PE, s],
     {j, {1, 2, 3}}];
  Bound[XiIota, OddTail, s] =
   Bound[XiIota, 3, s] + Sum[evi[j]^4/(1 - evi[j]^2) vi[j].Matrix[AiotabarNonRep, s].Vector[PE, s],
     {j, {1, 2, 3}}];
  Clear[v, vi, e, evi];
  , {s, {i, o}}]
```

We continue to compute the bound on the weighted coefficients:

```
In[4979]:= Do[
  Do[
   v[j] = EigenVectorB[j, s];
   vb[j] = EigenVectorBbar[j, s];
   vi[j] = EigenVectorBForPiota[j, s];
   e[j] = EigenValueB[j, s];
   eb[j] = EigenValueBbar[j, s];
   evi[j] = EigenValueBForPiota[j, s];
   , {j, {1, 2, 3}}];
  Bound[Xi, EvenTail, Delta, s] =
   Bound[Xi, 2, Delta, s] + Vector[hS, s].Transpose[Matrix[AiotaNonRep, s]].
```

```
    Sum[vb[j] * eb[j]^3 (2/(1 - eb[j]^2)^2 + 4/(1 - eb[j]^2)), {j, {1, 2, 3}}] +
   Sum[e[j]^3 (2/(1 - e[j]^2)^2 + 4/(1 - e[j]^2)) * v[j], {j, {1, 2, 3}}].
    (Matrix[H3, s].Vector[PE, s] + Matrix[AiotaNonRep, s].Vector[hE, s]) +
   2 Sum[v[j] 1/(1 - e[j]^2), {j, {1, 2, 3}}].Matrix[C1, s].
     Sum[vb[j] 1/(1 - eb[j]^2)^2, {j, {1, 2, 3}}] +
   2 Sum[v[j] 1/(1 - e[j]^2)^2, {j, {1, 2, 3}}].Matrix[C1, s].
     Sum[vb[j] 1/(1 - eb[j]^2), {j, {1, 2, 3}}] - 4 Vector[PS, s].Matrix[C1, s].Vector[PE, s] +
   2 Sum[v[j] e[j]/(1 - e[j]^2), {j, {1, 2, 3}}].Matrix[C2, s].
     Sum[vb[j] eb[j] (1/(1 - eb[j]^2)^2 + 1/(1 - eb[j]^2)), {j, {1, 2, 3}}] +
   2 Sum[v[j] e[j]/(1 - e[j]^2)^2, {j, {1, 2, 3}}].Matrix[C2, s].
     Sum[vb[j] eb[j] 1/(1 - eb[j]^2), {j, {1, 2, 3}}];

Bound[Xi, OddTail, Delta, s] =
  Bound[Xi, 3, Delta, s] + Vector[hS, s].Transpose[Matrix[AiotaNonRep, s]].
    Sum[vb[j] * eb[j]^4 (2/(1 - eb[j]^2)^2 + 5/(1 - eb[j]^2)), {j, {1, 2, 3}}] +
   Sum[e[j]^4 (2/(1 - e[j]^2)^2 + 5/(1 - e[j]^2)) * v[j], {j, {1, 2, 3}}].
    (Matrix[H3, s].Vector[PE, s] + Matrix[AiotaNonRep, s].Vector[hE, s]) +
   Sum[v[j] 2/(1 - e[j]^2)^2, {j, {1, 2, 3}}].
    (Matrix[C1, s].Matrix[Bbar, s] + Matrix[B, s].Matrix[C2, s]).
    Sum[vb[j] 1/(1 - eb[j]^2), {j, {1, 2, 3}}] +
   Sum[v[j] 1/(1 - e[j]^2), {j, {1, 2, 3}}].
    (Matrix[C1, s].Matrix[Bbar, s] + Matrix[B, s].Matrix[C2, s]).
    Sum[vb[j] (2/(1 - eb[j]^2)^2 + 1/(1 - eb[j]^2)), {j, {1, 2, 3}}] -
   5 Vector[PS, s].(Matrix[C1, s].Matrix[Bbar, s] + Matrix[B, s].Matrix[C2, s]).
     Vector[PE, s];
```

```
Bound[XiIota, EvenTail, Delta, ei, s] =
 Bound[XiIota, 2, Delta, ei, s] +
  Vector[hII, s].Sum[vb[j] * eb[j]^3 (2/(1 - eb[j]^2)^2 + 3/(1 - eb[j]^2)), {j, {1, 2, 3}}] +
  Sum[evi[j]^3 vi[j] (2/(1 - evi[j]^2)^2 + 3/(1 - evi[j]^2)), {j, {1, 2, 3}}].
   (Matrix[H3, s].Vector[PE, s] + Matrix[AiotaNonRep, s].Vector[hE, s]) +
  Sum[vi[j] (2 evi[j])/(1 - evi[j]^2)^2, {j, {1, 2, 3}}].
   (Matrix[C2, s].Matrix[Bbar, s] + Matrix[B, s].Matrix[C1, s]).
   Sum[vb[j] 1/(1 - eb[j]^2), {j, {1, 2, 3}}] +
  Sum[vi[j] evi[j]/(1 - evi[j]^2), {j, {1, 2, 3}}].
   (Matrix[C2, s].Matrix[Bbar, s] + Matrix[B, s].Matrix[C1, s]).
   Sum[vb[j] (2/(1 - eb[j]^2)^2 + 1/(1 - eb[j]^2)), {j, {1, 2, 3}}];
Bound[XiIota, OddTail, Delta, ei, s] =
 Bound[XiIota, 3, Delta, ei, s] +
  Vector[hII, s].Sum[vb[j] * eb[j]^4 (2/(1 - eb[j]^2)^2 + 4/(1 - eb[j]^2)), {j, {1, 2, 3}}] +
  Sum[evi[j]^4 (2/(1 - evi[j]^2)^2 + 4/(1 - evi[j]^2)) * vi[j], {j, {1, 2, 3}}].
   (Matrix[H2, s].Vector[PE, s] + Matrix[AiotaNonRep, s].Vector[hE, s]) +
  Sum[vi[j] (2 evi[j])/(1 - evi[j]^2)^2, {j, {1, 2, 3}}].Matrix[C2, s].
   Sum[vb[j] 1/(1 - eb[j]^2), {j, {1, 2, 3}}] +
  Sum[vi[j] evi[j]/(1 - evi[j]^2), {j, {1, 2, 3}}].Matrix[C2, s].
   Sum[vb[j] 2/(1 - eb[j]^2)^2, {j, {1, 2, 3}}] -
  4 Vector[Piota, s].Matrix[B, s].Matrix[C2, s].Vector[PE, s] +
  Sum[vi[j] (2 evi[j]^2)/(1 - evi[j]^2)^2, {j, {1, 2, 3}}].Matrix[C1, s].
   Sum[vb[j] 1/(1 - eb[j]^2), {j, {1, 2, 3}}] +
```

```
  Sum[vi[j] evi[j]^2/(1 - evi[j]^2), {j, {1, 2, 3}}].Matrix[C1, s].
   Sum[vb[j] (2/(1 - eb[j]^2)^2 + 2/(1 - eb[j]^2)), {j, {1, 2, 3}}];



Bound[XiIota, EvenTail, Delta, 0, s] =
 Bound[XiIota, 2, Delta, 0, s] +
  Sum[evi[j]^3/(1 - evi[j]^2) (2 + 4/(1 - evi[j]^2)) vi[j].Matrix[AiotabarNonRep, s].Vector[PE, s],
   {j, {1, 2, 3}}] +
  Vector[hII, s].Sum[vb[j] * eb[j]^3 (2/(1 - eb[j]^2)^2 + 4/(1 - eb[j]^2)), {j, {1, 2, 3}}] +
  Sum[evi[j]^3 vi[j] (2/(1 - evi[j]^2)^2 + 4/(1 - evi[j]^2)), {j, {1, 2, 3}}].
   (Matrix[H3, s].Vector[PE, s] + Matrix[AiotaNonRep, s].Vector[hE, s]) +
  Sum[vi[j] 2 evi[j]/(1 - evi[j]^2)^2, {j, {1, 2, 3}}].
   (Matrix[C2, s].Matrix[Bbar, s] + Matrix[B, s].Matrix[C1, s]).
   Sum[vb[j] 1/(1 - eb[j]^2), {j, {1, 2, 3}}] +
  Sum[vi[j] evi[j]/(1 - evi[j]^2), {j, {1, 2, 3}}].
   (Matrix[C2, s].Matrix[Bbar, s] + Matrix[B, s].Matrix[C1, s]).
   Sum[vb[j] (2/(1 - eb[j]^2)^2 + 4/(1 - eb[j]^2)), {j, {1, 2, 3}}];
Bound[XiIota, OddTail, Delta, 0, s] =
 Bound[XiIota, 3, Delta, 0, s] +
  Sum[evi[j]^4/(1 - evi[j]^2) (2 + 5/(1 - evi[j]^2)) vi[j].Matrix[AiotabarNonRep, s].Vector[PE, s],
   {j, {1, 2, 3}}] +
  Vector[hII, s].Sum[vb[j] * eb[j]^4 (2/(1 - eb[j]^2)^2 + 5/(1 - eb[j]^2)), {j, {1, 2, 3}}] +
  Sum[evi[j]^4 (2/(1 - evi[j]^2)^2 + 5/(1 - evi[j]^2)) * vi[j], {j, {1, 2, 3}}].
   (Matrix[H2, s].Vector[PE, s] + Matrix[AiotaNonRep, s].Vector[hE, s]) +
  Sum[vi[j] 2 evi[j]/(1 - evi[j]^2)^2, {j, {1, 2, 3}}].Matrix[C2, s].
   Sum[vb[j] 1/(1 - eb[j]^2), {j, {1, 2, 3}}] +
```

```
    Sum[vi[j] evi[j]/(1 - evi[j]^2), {j, {1, 2, 3}}].Matrix[C2, s].
     Sum[vb[j] (2/(1 - eb[j]^2)^2 + 1/(1 - eb[j]^2)), {j, {1, 2, 3}}] -
    5 Vector[Piota, s].Matrix[B, s].Matrix[C2, s].Vector[PE, s] +
    Sum[vi[j] (2 evi[j]^2)/(1 - evi[j]^2)^2, {j, {1, 2, 3}}].Matrix[C1, s].
     Sum[vb[j] 1/(1 - eb[j]^2), {j, {1, 2, 3}}] +
    Sum[vi[j] evi[j]^2/(1 - evi[j]^2), {j, {1, 2, 3}}].Matrix[C1, s].
     Sum[vb[j] (2/(1 - eb[j]^2)^2 + 3/(1 - eb[j]^2)), {j, {1, 2, 3}}];

  Clear[v, vi, vb, e, evi, eb];
  , {s, {i, o}}]
```

Summing the bounds

We compute the sum over all odd/even N as follows:

In[4980]:=
```
Do[
  Bound[Xi, Even, s] = Sum[Bound[Xi, t, s], {t, {0, EvenTail}}];
  Bound[Xi, Odd, s] = Sum[Bound[Xi, t, s], {t, {1, OddTail}}];
  Bound[Xi, Absolut, s] = Sum[Bound[Xi, t, s], {t, {Odd , Even}}];

  Bound[Xi, Even, Delta, s] = Sum[Bound[Xi, t, Delta, s], {t, {0, EvenTail}}];
  Bound[Xi, Odd, Delta, s] = Sum[Bound[Xi, t, Delta, s], {t, {1, OddTail}}];
  Bound[Xi, Absolut, Delta, s] = Sum[Bound[Xi, t, Delta, s], {t, {Odd , Even}}];



  Bound[XiIota, Even, s] = Sum[Bound[XiIota, t, s], {t, {0, EvenTail}}];
  Bound[XiIota, Odd, s] = Sum[Bound[XiIota, t, s], {t, {1, OddTail}}];
  Bound[XiIota, Absolut, s] = Sum[Bound[XiIota, t, s], {t, {Odd , Even}}];

  Do[
   Bound[XiIota, Even, Delta, type, s] =
    Sum[Bound[XiIota, t, Delta, type, s], {t, {0, EvenTail}}];
   Bound[XiIota, Odd, Delta, type, s] =
    Sum[Bound[XiIota, t, Delta, type, s], {t, {1, OddTail}}];
   Bound[XiIota, Absolut, Delta, type, s] =
    Sum[Bound[XiIota, t, Delta, type, s], {t, {Odd, Even}}];
   , {type, {ei, 0}}];

  Clear[type, t];
  , {s, {i, o}}]
```

The simplified rewrite of the NoBLE in (I)

In the preceeding section, we have computed all bounds required by Assumption 4.3 of (I). We use the methods provided in ***General.nb*** to compute the bounds on the simplified rewrite of (I), as derived in Appendix D of (I).

In[4981]:=
```
Do[
  mu[s] = Gamma1 / ((2 d - 1) cmu);
  mub[s] = z[s];
  mumin[s] = z[i] (1 - Max[ Bound[G, {1}, 3, s], Bound[G, {1}, 3, i] ]);

  beta[CPhi, Lower, s] = betaCPhiLow[d, mu[s], Bound[Xi, alpha, OneMinusZero, AtZero, s],
    Bound[XiIota, alphaI, 0, Atei, s]];
  beta[CPhi, Upper, s] = betaCPhiUp[d, mu[s], Bound[Xi, alpha, ZeroMinusOne, AtZero, s],
    Bound[XiIota, alphaII, 0, AtZero, s]];

  beta[af, Lower, s] = betaAfLow[d, mumin[s], mu[s],
    Bound[Psi, alphaI, OneMinusZero, AroundEi, s],
```

```
  Bound[Psi, alphaII, ZeroMinusOne, AroundZero, s], Bound[Pi, alpha, 0, s]];
beta[af, Upper, s] = betaAfUp[d, mu[s], Bound[Psi, alphaI, ZeroMinusOne, AroundEi, s],
  Bound[Psi, alphaII, OneMinusZero, AroundZero, s], Bound[Pi, alpha, lower, 0, s]];

beta[ap, s] = betaap[d, mu[s], Bound[Xi, alpha, OneMinusZero, AtEi, s],
  Bound[Xi, alpha, ZeroMinusOne, AtEi, s], Bound[XiIota, alphaI, 0, SumAroundei, s],
  Bound[XiIota, alphaII, 0, SumAroundZero, s]];

beta[PiHat, s] = betaPiHat[d, (2 d - 1)/(2 d) mub[s], Bound[XiIota, Even, s],
  Bound[Pi, 1, Lower, s]];
beta[PsiHat, s] = betaPsiHatLower[d, (2 d - 1)/(2 d) mubOverMu[s], Bound[Xi, Odd, s],
  Bound[Psi, lower, 0, s]];


beta[Rf, s] = betaRF[d, mu[s], mub[s], (2 d - 1)/(2 d) mubOverMu[s], (2 d - 1)/(2 d) mub[s],
  Bound[Xi, Absolut, s], Bound[Xi, EvenTail, s] + Bound[Xi, OddTail, s],
  Bound[XiIota, Absolut, s], Sum[Bound[XiIota, t, s], {t, {Odd , EvenTail}}],
  Bound[Psi, RI, 0, s] + Bound[Psi, RI, 1, s],
  Bound[Psi, RII, 0, s] + Bound[Psi, RII, 1, s], Bound[Pi, R, 0, s]];


beta[Rp, s] = betaRp[d, mu[s], (2 d - 1)/(2 d) mubOverMu[s], (2 d - 1)/(2 d) mub[s],
  Bound[Xi, Absolut, s], Bound[Xi, R, 0, s] + Bound[Xi, R, 1, s],
  Bound[Xi, EvenTail, s] + Bound[Xi, OddTail, s], Bound[XiIota, Absolut, s],
  Sum[Bound[XiIota, t, s], {t, {Odd , EvenTail}}], Bound[XiIota, RI, 0, s],
  Bound[XiIota, RII, 0, s]];

beta[Rp, Delta, s] = betaRpDelta[d, mu[s], mubOverMu[s], mub[s],
  Bound[Xi, Absolut, s], Bound[Xi, Absolut, Delta, s],
  Bound[Xi, R, 0, s] + Bound[Xi, R, 1, s],
  Sum[Bound[Xi, t, Delta, s], {t, {OddTail , EvenTail}}], Bound[XiIota, Absolut, s],
  Bound[XiIota, Absolut, Delta, ei, s], Bound[XiIota, Absolut, Delta, 0, s],
  Sum[Bound[XiIota, t, Delta, ei, s], {t, {Odd, EvenTail}}],
  Sum[Bound[XiIota, t, Delta, 0, s], {t, {Odd, EvenTail}}],
  Bound[XiIota, RI, 0, Delta, ei, s], Bound[XiIota, RII, 0, Delta, 0, s]];


beta[Rf, abs, Delta, s] = betaRfDelta[d, mu[s], mubOverMu[s], mub[s],
  Bound[Xi, Absolut, s], Bound[Xi, Absolut, Delta, s],
  Bound[Xi, EvenTail, s] + Bound[Xi, OddTail, s],
```

```
  Sum[Bound[Xi, t, Delta, s], {t, {OddTail , EvenTail}}],
  Bound[Psi, RI, 0, Delta, s] + Bound[Psi, RI, 1, Delta, s],
  Bound[Psi, RII, 0, Delta, s] + Bound[Psi, RII, 1, Delta, s],
  Bound[XiIota, Absolut, s], Bound[XiIota, Absolut, Delta, ei, s],
  Bound[XiIota, Absolut, Delta, 0, s], Sum[Bound[XiIota, t, s], {t, {Odd, EvenTail}}],
  Sum[Bound[XiIota, t, Delta, ei, s], {t, {Odd, EvenTail}}],
  Bound[Pi, R, 0, Delta, eiek, s]];

beta[Rf, Lower, Delta, s] = betaRfDeltaLower[d, mu[s], mubOverMu[s], mub[s],
  Bound[Xi, Absolut, s], Bound[Xi, Odd, s], Bound[Xi, Even, s],
  Bound[Xi, Absolut, Delta, s], Bound[Xi, Odd, Delta, s], Bound[Xi, Even, Delta, s],
  Bound[Xi, OddTail, s], Bound[Xi, OddTail, Delta, s], Bound[Xi, EvenTail, Delta, s],
  Bound[Psi, RI, 1, Delta, s], Bound[Psi, RII, 0, Delta, s], Bound[XiIota, Absolut, s],
  Bound[XiIota, Odd, s], Bound[XiIota, Even, s], Bound[XiIota, Absolut, Delta, ei, s],
  Bound[XiIota, Odd, Delta, ei, s], Bound[XiIota, Even, Delta, ei, s],
  Bound[XiIota, Absolut, Delta, 0, s], Bound[XiIota, Odd, Delta, 0, s],
  Bound[XiIota, Odd, Delta, 0, s], Bound[XiIota, EvenTail, s],
  Bound[XiIota, EvenTail, Delta, ei, s], Bound[Pi, R, 0, Delta, eiek, s]];
, {s, {i, o}}]
```

Improvement of Bounds

In this section, we implement the computations of Section 3 of (I) to verify whether we can conclude from $f_i(z) \leq \Gamma_i$ that $f_i(z) < \Gamma_i - \epsilon$. The sufficient condition for this to succeed is stated in Definition 2.9 of (I). We check the conditions one line at a time.

Technical condition

All the conditions below are necessary for our analysis to be valid. However, numerically, they turn out to always be true, in the sense that other conditions (most likely f_2) are more likely to fail before these do.

```
In[4982]:= Do[
  TechCondition[I, s] = (beta[CPhi, Lower, s] - beta[ap, s] - beta[Rp, s]) > 0;
  (* Condition in Assumption 2.7, stating that numerator of Gz =Φ̂(k) is positive *)
  TechCondition[II, s] = (beta[af, Lower, s] - beta[Rf, Lower, Delta, s]) > 0;
  (* Condition in Assumption 2.7, stating that the denominator of Ĝz(k) is positive,
  also neccesary to ensure that f2 is well define *)
  TechCondition[III, s] = ((2 d - 1) mub[s] / (1 - mu[s]) Bound[XiIota, Absolut, s]) < 1;
  (* Condition (4.33) of (I), which is necessary to use the geometric series *)
  TechCondition[s] = TechCondition[I, s] && TechCondition[II, s] && TechCondition[III, s]
  , {s, {i, o}}]
```

Improvement of f_1

We continue to implement the bounds derived in Section 3.1 of (I):

In[4983]:=
```
Do[
 boundF1[part1, s] = mubOverMu[s] (1 + beta[PiHat, s]) / (1 - (2 d)/(2 d - 1) beta[PsiHat, s]);
 boundF1[part2, s] = cmu (1 + beta[PiHat, s]) / (1 - (2 d)/(2 d - 1) beta[PsiHat, s]);
 boundF1[s] = Max[boundF1[part1, s], boundF1[part2, s]];

 , {s, {i, o}}]
```

Improvement of f_2

Here we implement the bounds derived in Section 3.2 of (I):

In[4984]:=
```
Do[
 boundF2[s] = (2 d - 1)/(2 d - 2) (beta[CPhi, Upper, s] + beta[ap, s] + beta[Rp, s])/(beta[af, Lower, s] + beta[Rf, Lower, Delta, s]);
 , {s, {i, o}}]
```

Improvement of f_3

First, we prepare the bound and compute the value at z_I:

In[4985]:=
```
(* The values we compare with *)
const[1] = c[1, 6, 0] ; (*c1,6,{0}*)
const[2] = c[0, 0, 1]; (*c0,0,Z/{0}*)
const[3] = c[1, 0, 1]; (*c1,0,Z/{0}*)
const[4] = c[1, 1, 1]; (*c1,1,Z/{0}*)
const[5] = c[1, 2, 1] ; (*c1,2,Z/{0}*)
const[6] = c[1, 3, 1]; (*c1,3,Z/{0}*)

(* Initial bounds as in Section 3.3.3 of (I), obtained by pure SRW computation,
Methods provided in the General.nb *)
boundF3[1, i] = BoundFThreeInital[d, 1, 6, 1, {{0}} ];
boundF3[2, i] = BoundFThreeInital[d, 0, 0, 1, {{1}, {2}, {0, 1}} ];
boundF3[3, i] = BoundFThreeInital[d, 1, 0, 1, {{1}, {2}, {0, 1}} ];
boundF3[4, i] = BoundFThreeInital[d, 1, 1, 1, {{1}, {2}, {0, 1}} ];
boundF3[5, i] = BoundFThreeInital[d, 1, 2, 1, {{1}, {2}, {0, 1}} ];
boundF3[6, i] = BoundFThreeInital[d, 1, 3, 1, {{1}, {2}, {0, 1}} ];
```

In[4997]:=
```
(* The bounds summarized in (3.80)of (I),
used methods are implemented in General.nb*)
BoundFThreeBound[n_, l_, vs_] :=
  BoundFThree[d, n, l, vs, (2 d - 2)/(2 d - 1) Gamma2, beta[CPhi, Upper, o], beta[af, Lower, o],
   beta[af, Upper, o], beta[ap, o], beta[Rp, o], beta[Rf, o], beta[Rf, abs, Delta, o],
   beta[Rp, Delta, o], 1 / (beta[af, Lower, o] + beta[Rf, Lower, Delta, o])];
```

For the part of $f_3$with $S = \{0\}$, $n = 1$, $l = 6$, we use $H^{1,6}(0) = H^{1,5}(e)$ for the bound

```
In[4998]:= boundF3[1, o] = Min[BoundFThreeBound[1, 6, {{0}}], BoundFThreeBound[1, 5, {{1}}]];
```

For $S = Z^d / \{0\}$ and $n = 0$ and $l = 0$ we use

$$\sup_y \| y \|_2^2 G_p(y) \leq \max \{G_p(e), 2 G_p(e_1 + e_2), 4 G_p(2 e), \sup_{y: |y|>2} \| y \|_2^2 G_p(y)\}$$

```
In[4999]:= boundF3[2, o] = If[ValueQ[T[2, 0, {3}]]
     , Max[{Bound[G, {1}, 1, o], 2 Bound[G, {2}, 2, o], 4 Bound[G, {0, 1}, 2, o],
       BoundFThreeBound[0, 0, {{3}, {1, 1}, {0, 0, 1}}]}],
     Max[{Bound[G, {1}, 1, o], 2 Bound[G, {2}, 2, o], 4 Bound[G, {0, 1}, 2, o],
       BoundFThreeBound[0, 0, {{2}, {0, 1}}]}]];
```

For $S = Z^d / \{0\}$ and $n = 1$ and $l = 2, 3$ we use just use the pre-implemented function, that compute (3.96) of (I)

```
In[5000]:= boundF3[5, o] = BoundFThreeBound[1, 2, {{1}, {0, 1}, {2}}];
           boundF3[6, o] = BoundFThreeBound[1, 3, {{1}, {0, 1}, {2}}];
```

For $S = Z^d / \{0\}$ and $n = 1$ and l=1, we note that

$$H^{1,1}(e) \leq 2 d p H^{1,2}(e) + \frac{1}{2 d} \sup_y \sum_e \| y - e \|_2^2 G_z(y - e) = 2 d p H^{1,2}(e) + \frac{1}{2 d} (2 (2 d - 2) G_p(e_1 + e_2) + 4 G_p(2 e))$$
$$\sup_{y \neq 0} H^{1,1}(y) = \max (H^{1,1}(e), \sup_{y:|y| \geq 2} H^{1,1}(y))$$

to obtain a better estimate

```
In[5002]:= tmpNeighborContributionOneD =
    2 d z[o] BoundFThreeBound[1, 2, {{1}}] +
     1/(2 d) ((2 d - 2) 2 Bound[G, {2}, 2, o] + 4 Bound[G, {0, 1}, 2, o]);
   boundF3[4, o] = Max[tmpNeighborContributionOneD, BoundFThreeBound[1, 1, {{0, 1}, {2}}]];
```

For $S = Z^d / \{0\}$ and $n = 1$ and l=0, we extend the idea of l=1 once more

$$H^{1,0}(x) \leq 2 d p H^{1,1}(x) + \| x \|_2^2 G_p(x)$$
$$\sup_{y \neq 0} H^{1,0}(y) = \max (H^{1,0}(e), H^{1,0}(e_1 + e_2), H^{1,0}(2 e), \sup_{y:|y|>2} H^{1,1}(y))$$

```
In[5004]:= boundF3[3, o] = If[ValueQ[T[2, 0, {3}]],
     Max[2 d z[o] tmpNeighborContributionOneD + Bound[G, {1}, 1, o],
      BoundFThreeBound[1, 1, {{2}}] + 2 Bound[G, {2}, 2, o],
      BoundFThreeBound[1, 1, {{0, 1}}] + 2 Bound[G, {0, 1}, 2, o],
      BoundFThreeBound[1, 0, {{3}, {1, 1}, {0, 0, 1}}]],
     Max[2 d z[o] tmpNeighborContributionOneD + Bound[G, {1}, 1, o],
      BoundFThreeBound[1, 1, {{2}}] + 2 Bound[G, {2}, 2, o],
      BoundFThreeBound[1, 1, {{0, 1}}] + 2 Bound[G, {0, 1}, 2, o],
      BoundFThreeBound[1, 0, {{2}, {0, 1}}]]];
```

```
In[5005]:= BoundsF3Table = Table[boundF3[j, s] / const[j], {s, {i, o}}, {j, 1, 6}];
boundF3[i] = Max[BoundsF3Table[[1]]] * Gamma3;
boundF3[o] = Max[BoundsF3Table[[2]]] * Gamma3;
```

Results

Preparation of output

```
In[5052]:= Do[
   SuccesF[1, s] = boundF1[s] < Gamma1;
   SuccesF[2, s] = (boundF2[s] < Gamma2) && (boundF2[s] > 0);
   SuccesF[3, s] = (boundF3[s] < Gamma3) && (boundF3[s] > 0);
   Succes[s] = SuccesF[1, s] && SuccesF[2, s] && SuccesF[3, s] && TechCondition[s];
   , {s, {i, o}}];
Succes[overall] = Succes[i] && Succes[o];
overAllStatement = "The statement that the bootstrap was succesful is "
   If[Succes[overall], Style[TextString[Succes[overall]], Bold, Green],
    Style[TextString[Succes[overall]], Bold, Red]];
```

```
In[5010]:= bubbles = {Graphics[{Green, Disk[{0, 0}, 0.8]}, ImageSize -> 30],
    Graphics[{Red, Disk[{0, 0}, 0.8]}, ImageSize -> 40]};


Do[
 CoefficientboundsTable[s] =
  {{Quantity, "Ξ^Zero", "Ξ^One", "Ξ^Two", "Ξ^Three", "Ξ^Even,>3", "Ξ^Odd,>3"},
   {Text[Bound for Ξ̂], Bound[Xi, 0, s], Bound[Xi, 1, s] , Bound[Xi, 2, s],
    Bound[Xi, 3, s] , Bound[Xi, EvenTail, s] - Bound[Xi, 2, s],
    Bound[Xi, OddTail, s] - Bound[Xi, 3, s]},
   {Text[Bound for Ξ̂^ι], Bound[XiIota, 0, s], Bound[XiIota, 1, s] , Bound[XiIota, 2, s],
    Bound[XiIota, 3, s] , Bound[XiIota, EvenTail, s] - Bound[XiIota, 2, s],
    Bound[XiIota, OddTail, s] - Bound[XiIota, 3, s]},
   {Text[Ξ̂ "||x||_2^2" ], Bound[Xi, 0, Delta, s], Bound[Xi, 1, Delta, s],
    Bound[Xi, 2, Delta, s], Bound[Xi, 3, Delta, s],
    Bound[Xi, EvenTail, Delta, s] - Bound[Xi, 2, Delta, s],
    Bound[Xi, OddTail, Delta, s] - Bound[Xi, 3, Delta, s]},
   {Text[Ξ^ι "||x-e_ι||_2^2"] , Bound[XiIota, 0, Delta, ei, s],
    Bound[XiIota, 1, Delta, ei, s] , Bound[XiIota, 2, Delta, ei, s],
    Bound[XiIota, 3, Delta, ei, s],
    Bound[XiIota, EvenTail, Delta, ei, s] - Bound[XiIota, 2, Delta, ei, s],
    Bound[XiIota, OddTail, Delta, ei, s] - Bound[XiIota, 3, Delta, ei, s]},
   {Text[Ξ^ι "||x||_2^2"] , Bound[XiIota, 0, Delta, 0, s], Bound[XiIota, 1, Delta, 0, s] ,
    Bound[XiIota, 2, Delta, 0, s], Bound[XiIota, 3, Delta, 0, s],
    Bound[XiIota, EvenTail, Delta, 0, s] - Bound[XiIota, 2, Delta, 0, s],
```

```
    Bound[XiIota, OddTail, Delta, 0, s] - Bound[XiIota, 3, Delta, 0, s]}};
TableCoefficients[s] =
  Labeled[Grid[CoefficientboundsTable[s], Alignment → {Center}, Frame → True,
     Dividers → {{2 → True, -1 → True}, {2 → True}}, ItemStyle → {1 → Bold, 1 → Bold},
     Background → {{None}, {GrayLevel[0.9]}, {None}}],
    Style["Bound on the coefficients in dimension " <> TextString[d], Bold], Top] //
   Text;



CoefficientSplitsBoundsI[s] =
  {{Quantity, "Ξ_R^Zero", "Ψ_RI^Zero", "Ψ_RII^Zero", "Ξ_R^One", "Ψ_RI^One", "Ψ_RII^One"},
   {Text[Abs Bound], Bound[Xi, R, 0, s], Bound[Psi, RI, 0, s], Bound[Psi, RII, 0, s],
    Bound[Xi, R, 1, s] , Bound[Psi, RI, 1, s], Bound[Psi, RII, 1, s]},
   {Text[Ξ̂ "||x||_2^2" ], Bound[Xi, R, 0, Delta, s], Bound[Psi, RI, 0, Delta, s],
    Bound[Psi, RII, 0, Delta, s], Bound[Xi, R, 1, Delta, s] ,
    Bound[Psi, RI, 1, Delta, s], Bound[Psi, RII, 1, Delta, s]}};
TableCoefficientSplitI[s] =
  Labeled[Grid[CoefficientSplitsBoundsI[s], Alignment → {Center}, Frame → True,
     Dividers → {{2 → True, -1 → True}, {2 → True}}, ItemStyle → {1 → Bold, 1 → Bold},
     Background → {{None}, {GrayLevel[0.9]}, {None}}],
    Style["Bound on the remainder term of the split of Ξ and Ψ." Text[d], Bold],
    Top] // Text;

CoefficientSplitsBoundsII[s] =
  {{Quantity, " Ξ̂_alpha^zero(0)-Ξ̂_alpha^One(0)", " Ξ̂_alpha^zero(e_1)-Ξ̂_alpha^One(e_1)", "∑Ψ_alphaI^Zero-Ψ_alphaI^One",
    "∑Ψ_alphaII^Zero-Ψ_alphaII^One"}, {Text[Lower Bound], Bound[Xi, alpha, OneMinusZero, AtZero, s],
    Bound[Xi, alpha, OneMinusZero, AtEi, s],
    Bound[Psi, alphaI, OneMinusZero, AroundEi, s],
    Bound[Psi, alphaII, OneMinusZero, AroundZero, s]},
   {Text[Upper Bound ], Bound[Xi, alpha, ZeroMinusOne, AtZero, s],
    Bound[Xi, alpha, ZeroMinusOne, AtEi, s],
    Bound[Psi, alphaI, ZeroMinusOne, AroundEi, s],
    Bound[Psi, alphaII, ZeroMinusOne, AroundZero, s]}};
TableCoefficientSplitII[s] =
  Labeled[Grid[CoefficientSplitsBoundsII[s], Alignment → {Center}, Frame → True,
     Dividers → {{2 → True, 3 → True, 4 → True, 5 → True, -1 → True}, {2 → True}},
     ItemStyle → {1 → Bold, 1 → Bold}, Background → {{None}, {GrayLevel[0.9]}, {None}}],
    Style["Bound on the difference of explicit terms in dimension " <> TextString[d],
     Bold], Top] // Text;

CoefficientSplitsBoundsIII[s] =
  {{"Quantity", "-", "-", " Ξ̂^ι(0)", "∑_x |x-e_ι|_2^2 Ξ^ι(x)", "∑_x |x|_2^2 Ξ^ι(x)"},
   {"Bound", "-", "-", ScientificForm[Bound[XiIota, 0, s], 5],
```

```
      ScientificForm[Bound[XiIota, 0, Delta, ei, s], 5],
      ScientificForm[Bound[XiIota, 0, Delta, 0, s], 5]},
     {"Split I", " Ξ^ι_α,I(e_ι)", "∑_κ Ξ^ι_α,I(e_ι+e_κ)", " Ξ̂^ι_RI(0)", "∑_x |x-e_ι|_2^2 Ξ̂^ι_R,I", "-"},
     {"Bound", ScientificForm[Bound[XiIota, alphaI, 0, Atei, s], 5],
      ScientificForm[Bound[XiIota, alphaI, 0, SumAroundei, s], 5],
      ScientificForm[Bound[XiIota, RI, 0, s], 5],
      ScientificForm[Bound[XiIota, RI, 0, Delta, ei, s], 5], "-"},
     {"Split II", "Ξ^ι_α,II(0)", "∑_κ Ξ^ι_α,II(e_κ)", "Ξ̂^ι_R,II(0)", "-", "∑_x |x|_2^2 Ξ^ι_R,II(x)"},
     {"Bound", ScientificForm[Bound[XiIota, alphaII, 0, AtZero, s], 5],
      ScientificForm[Bound[XiIota, alphaII, 0, SumAroundZero, s], 5],
      ScientificForm[Bound[XiIota, RII, 0, s], 5], "-",
      ScientificForm[Bound[XiIota, RII, 0, Delta, 0, s], 5]}};
   TableCoefficientSplitIII[s] =
    Labeled[Grid[CoefficientSplitsBoundsIII[s], Alignment → {Center}, Frame → True,
       Dividers → {{2 → True, 4 → True, -1 → True}, {2 → True}},
       ItemStyle → {1 → Bold, 1 → Bold}, Background → {{None}, {{GrayLevel[0.9], None}}}],
      Style["Bound on split of the coefficient Ξ^(0),ι in dimension " Text[d], Bold],
      Top] // Text;
   , {s, {i, o}}]

SimpleNotationBoundsPart1 =
  {{Quantity, "a_F-Lower", "a_F-Upper", "|a_Φ|", "c_Φ-Lower", "c_Φ-Upper", Π, Ψ},
   {"Bound i" , beta[af, Lower, i], beta[af, Upper, i], beta[ap, i],
    beta[CPhi, Lower, i], beta[CPhi, Upper, i], NumberForm[beta[PiHat, i], 7],
    NumberForm[beta[PsiHat, i], 7]},
   {"Bound o" , beta[af, Lower, o], beta[af, Upper, o], beta[ap, o],
    beta[CPhi, Lower, o], beta[CPhi, Upper, o], NumberForm[beta[PiHat, o], 7],
    NumberForm[beta[PsiHat, o], 7]}};

TableSimpleNotation1 =
  Labeled[Grid[SimpleNotationBoundsPart1, Alignment → {Center}, Frame → True,
     Dividers → {{2 → True, -1 → True}, {2 → True}},
     ItemStyle → {1 → Bold, 1 → Bold}, Background → {{None}, {GrayLevel[0.9]}, {None}}],
    Style["Bound on the simplified rewrite in dimension " <> TextString[d], Bold],
    Top] // Text;

SimpleNotationBoundsPart2 =
  {{Quantity, "|R_F|", "|R_Φ|", "x_2^2R_F-lower", "|x_2^2R_F|", "|x_2^2R_Φ|"},
   {"Bound i", beta[Rf, i], beta[Rp, i], beta[Rf, Lower, Delta, i],
    beta[Rf, abs, Delta, i], beta[Rp, Delta, i]},
   {"Bound o", beta[Rf, o], beta[Rp, o], beta[Rf, Lower, Delta, o],
    beta[Rf, abs, Delta, o], beta[Rp, Delta, o]}};
```

```
TableSimpleNotation2 =
  Labeled[Grid[SimpleNotationBoundsPart2, Alignment → {Center}, Frame → True,
     Dividers → {{2 → True, 4 → True, -1 → True}, {2 → True}},
     ItemStyle → {1 → Bold, 1 → Bold}, Background → {{None}, {GrayLevel[0.9]}, {None}}],
    Style["Bound on remainder of the rewrite in dimension " <> TextString[d], Bold],
    Top] // Text;

ContentCheckf1f2 = {{Bounds, "f1(zI)", "f2(zI)", "f3(zI)", "f1", "f2", "f3"},
   {"Assumed bound Γi", NumberForm[Gamma1, 5], NumberForm[Gamma2, 5],
    NumberForm[Gamma3, 5], NumberForm[Gamma1, 9], NumberForm[Gamma2, 9],
    NumberForm[Gamma3, 8] }, {"Concluded bound", NumberForm[boundF1[i], 5],
    NumberForm[boundF2[i], 5], NumberForm[boundF3[i], 5], NumberForm[boundF1[o], 9],
    NumberForm[boundF2[o], 9], NumberForm[boundF3[o], 6]}, { "Comparison",
    If[SuccesF[1, i], bubbles[[1]], bubbles[[2]]],
    If[SuccesF[2, i], bubbles[[1]], bubbles[[2]]],
    If[SuccesF[3, i], bubbles[[1]], bubbles[[2]]],
    If[SuccesF[1, o], bubbles[[1]], bubbles[[2]]],
    If[SuccesF[2, o], bubbles[[1]], bubbles[[2]]],
    If[SuccesF[3, o], bubbles[[1]], bubbles[[2]]]}};

TableCheckf1f2 =
  Labeled[Grid[ContentCheckf1f2, Alignment → {Center}, Frame → True,
     Dividers → {{2 → True, 5 → True, -1 → True}, {2 → True}},
     ItemStyle → {1 → Bold, 1 → Bold}, Background → {{None}, {GrayLevel[0.9]}, {None}}],
    Style["Result of the improvement of bounds in dimension " <> TextString[d], Bold],
    Top] // Text;

Do[
 tableCheckf3[s] =
   {{Bounds, "F3-1,6,{0}", "F3-0,0,x≠0", "F3-1,0,x≠0", "F3-1,1,x≠0",
     "F3-1,2,x≠0", "F3-1,3,x≠0"},
    {"Assumed bound", const[1] * Gamma3 , const[2] * Gamma3, const[3] * Gamma3,
     const[4] * Gamma3, const[5] * Gamma3, const[6] * Gamma3 },
    {"Concluded bound", boundF3[1, s], boundF3[2, s], boundF3[3, s], boundF3[4, s],
     boundF3[5, s], boundF3[6, s]}, { "Comparison",
     If[boundF3[1, s] < const[1] * Gamma3, bubbles[[1]], bubbles[[2]]],
     If[boundF3[2, s] < const[2] * Gamma3, bubbles[[1]], bubbles[[2]]],
     If[boundF3[3, s] < const[3] * Gamma3, bubbles[[1]], bubbles[[2]]],
     If[boundF3[4, s] < const[4] * Gamma3, bubbles[[1]], bubbles[[2]]],
     If[boundF3[5, s] < const[5] * Gamma3, bubbles[[1]], bubbles[[2]]],
     If[boundF3[6, s] < const[6] * Gamma3, bubbles[[1]], bubbles[[2]]]}};
 , {s, {i, o}}]
TableCheckf3 =
```

```
Labeled[Grid[tableCheckf3[o], Alignment → {Center}, Frame → True,
   Dividers → {{2 → True, -1 → True}, {2 → True}}, ItemStyle → {1 → Bold, 1 → Bold},
   Background → {{None}, {GrayLevel[0.9]}, {None}}],
  Style["Detailed result of the improvement of bounds for f₃ in dimension  " <>
    TextString[d], Bold], Top] // Text;
```

Simple output

In[5055]:= `overAllStatement`

Out[5055]= The statement that the bootstrap was succesful is **True**

If this succeeds, then the infrared bound, stated in Theorem 1.1 of both (I) and (II), holds with

In[5056]:=
```
"τ̂z(k)[1-D̂(k)] ≤ " <> TextString[Ceiling[(2 d - 2)/(2 d - 1) Gamma2, 0.001]]
"A₁(d) = " <>
 TextString[
  Ceiling[g[o] (beta[CPhi, Upper, o] + beta[ap, o] + beta[Rp, o])
    Max[1 / (beta[af, Lower, o] + beta[Rf, Lower, Delta, o]),
     1 / (beta[CPhi, Lower, o] - beta[ap, o] - beta[Rp, o])], 0.001]]
```

Out[5056]= $\hat{\tau}_z(k)[1-\hat{D}(k)] \leq 1.024$

Out[5057]= $A_1(d) = 3.069$

Further, we have proven that:

In[5059]:=
```
Print["pc ≤ ", Ceiling[1/(2 d - 1) Gamma1, 0.0001]]
```

$p_c \leq 0.0483$

The bound on the bootstrap functions

In[5024]:=
```
TableCheckf1f2
TableCheckf3
```

Out[5024]=

Result of the improvement of bounds in dimension 11

Bounds	$f_1(z_I)$	$f_2(z_I)$	$f_3(z_I)$	f_1	f_2	f_3
Assumed bound Γ_i	1.0131	1.0751	1	1.01306	1.07513	1
Concluded bound	1.0112	1.0505	0.7656	1.01305907	1.07512348	0.999971
Comparison	●	●	●	●	●	●

Out[5025]=

Detailed result of the improvement of bounds for f_3 in dimension 11

Bounds	F3−1,6,{0}	F3−0,0,x≠0	F3−1,0,x≠0	F3−1,1,x≠0	F3−1,2,x≠0	F3−1,3,x≠0
Assumed bound	0.008714	0.0661	0.108	0.05352	0.0404	0.022
Concluded bound	0.00871352	0.0660248	0.107997	0.0535177	0.040359	0.0219457
Comparison	●	●	●	●	●	●

Individual bounds

In[5026]:=
```
TableCoefficients[o]
TableSimpleNotation1
TableSimpleNotation2
TableCoefficientSplitI[o]
TableCoefficientSplitII[o]
TableCoefficientSplitIII[o]
```

Out[5026]=

Bound on the coefficients in dimension 11

Quantity	Ξ^{Zero}	Ξ^{One}	Ξ^{Two}	Ξ^{Three}	$\Xi^{Even,>3}$	$\Xi^{Odd,>3}$
Bound for $\hat{\Xi}$	0.00654178	0.0117727	0.0011801	0.0000844158	7.44395×10^{-6}	5.42936×10^{-7}
Bound for $\hat{\Xi}^{\iota}$	0.00297681	0.000854426	0.0000752601	5.42861×10^{-6}	4.78248×10^{-7}	3.48816×10^{-8}
$\|\|x\|\|_2^2\,\hat{\Xi}$	0.0119436	0.0319702	0.0116095	0.00224654	0.000445473	0.000045248
$\|\|x-e_\iota\|\|_2^2\,\Xi^{\iota}$	0.0000353226	0.00224566	0.000767694	0.000139929	0.0000212803	9.23974×10^{-6}
$\|\|x\|\|_2^2\,\Xi^{\iota}$	0.00301213	0.00620017	0.00138006	0.000208432	0.0000330616	0.0000110198

Out[5027]=

Bound on the simplified rewrite in dimension 11

Quantity	a_F−Lower	a_F−Upper	$\|a_\Phi\|$	c_Φ−Lower	c_Φ−Upper	Π	Ψ
Bound i	1.04403	1.05817	0.00093734	0.99703	1.	0.002757146	0.005161960
Bound o	1.04365	1.05819	0.00103949	0.996863	1.	0.002984297	0.006687571

Out[5028]=

Bound on remainder of the rewrite in dimension 11

Quantity	$\|R_F\|$	$\|R_\Phi\|$	$x_2^2 R_F$−lower	$\|x_2^2 R_F\|$	$\|x_2^2 R_\Phi\|$
Bound i	0.0262924	0.0120136	−0.0315679	0.0562729	0.0184352
Bound o	0.0301527	0.0144077	−0.0519316	0.0918993	0.030509

Out[5029]=

Bound on the remainder term of the split of Ξ and Ψ. 11

Quantity	Ξ_R^{Zero}	Ψ_{RI}^{Zero}	Ψ_{RII}^{Zero}	Ξ_R^{One}	Ψ_{RI}^{One}	Ψ_{RII}^{One}
Abs Bound	0.00337628	0.00593187	0.00307845	0.00860722	0.0109721	0.00824036
$\|\|x\|\|_2^2\,\hat{\Xi}$	0.0103742	0.0185402	0.00945904	0.0319702	0.0438727	0.0320651

Out[5030]=

Bound on the difference of explicit terms in dimension 11

Quantity	$\hat{\Xi}_{alpha}^{zero}(0)-\hat{\Xi}_{alpha}^{One}(0)$	$\hat{\Xi}_{alpha}^{zero}(e_1)-\hat{\Xi}_{alpha}^{One}(e_1)$	$\sum\Psi_{alphaI}^{Zero}-\Psi_{alphaI}^{One}$	$\sum\Psi_{alphaII}^{Zero}-\Psi_{alphaII}^{One}$
Bound Lower	0	0.0000379709	0.000184491	0.000815864
Bound Upper	0	0.0000403919	0.0000842667	0.000863855

Out[5031]=

Bound on split of the coefficient $\Xi^{(0),\iota}$ in dimension 11

Quantity	–	–	$\hat{\Xi}^{\iota}(0)$	$\sum_x \|x-e_\iota\|_2^2\,\Xi^{\iota}(x)$	$\sum_x \|x\|_2^2\,\Xi^{\iota}(x)$
Bound	–	–	2.9768×10^{-3}	3.5323×10^{-5}	3.0121×10^{-3}
Split I	$\Xi_{\alpha,I}^{\iota}(e_\iota)$	$\sum_\kappa \Xi_{\alpha,I}^{\iota}(e_\iota+e_\kappa)$	$\hat{\Xi}_{RI}^{\iota}(0)$	$\sum_x \|x-e_\iota\|_2^2\,\hat{\Xi}_{R,I}^{\iota}$	–
Bound	2.9575×10^{-3}	8.8608×10^{-6}	9.9852×10^{-6}	3.0681×10^{-5}	–
Split II	$\Xi_{\alpha,II}^{\iota}(0)$	$\sum_\kappa \Xi_{\alpha,II}^{\iota}(e_\kappa)$	$\hat{\Xi}_{R,II}^{\iota}(0)$	–	$\sum_x \|x\|_2^2\,\Xi_{R,II}^{\iota}(x)$
Bound	0	2.9575×10^{-3}	1.9347×10^{-5}	–	5.467×10^{-5}

Number required for the proof of Section 7 (II)

Here we check whether the conditions necessary to prove the x-space asymptotics of the two-point function are valid. We rely on the conditions given to us by Takashi Hara (private communication), and we rely on his work.

Simple condition by Hara

We first check the basic condition already stated in "Decay of correlations in nearest-neighbor self-avoiding walk, percolation, lattice trees and animals" by T. Hara. This condition fails in small dimensions.

In[5032]:=
```
Tbar = Max[((2 d - 2)/(2 d - 1) Gamma2)^3 Ivalue[3, 0, {0}] - 1, ((2 d - 2)/(2 d - 1) Gamma2)^3 K[3, 0, {1}]]
(* T̄^(0,0) = sup_x τ^*3 (x) - δ_0,x *)
4 d z[o] (Tbar + 1/(2 d)) (1 + Tbar) (* Term in (2.6) *)
4 d z[o] (Tbar + 1/(2 d)) (1 + Tbar) < 1 (* that is supposed to be smaller than 1*)
```

Out[5032]= 0.53561

Out[5033]= 1.89397

Out[5034]= False

Improved condition by Hara

As the former condition uses to fail in small dimension we ask Takashi Hara for a sharper condition, which would be

In[5035]:=
```
Tbar = Max[((2 d - 2)/(2 d - 1) Gamma2)^3 Ivalue[3, 0, {0}] - 1, ((2 d - 2)/(2 d - 1) Gamma2)^3 K[3, 0, {1}]]
(* T̄^(0,0) = sup_x τ^*3 (x) - δ_0,x *)
2 d z[o] (Tbar + 1/(2 d)) (1 + 2 Tbar) (* Term in (2.6) *)
2 d z[o] (Tbar + 1/(2 d)) (1 + 2 Tbar) < 1
```

Out[5035]= 0.53561

Out[5036]= 1.27729

Out[5037]= False

Using a better estimate on the involved diagram we conclude the following even sparper condition:

$$2\,d\,p\left(\tau^{*3}*D\right)(x) = 2\,d\,p\,D(x) + \sum_{r=1}^{3} 2\,d\,p\left(\tau^{*r}*D^{*2}\right)(x) \leq p\,\delta_{|x|=1} + 2\,d\,p\sum_{r=1}^{3}\left(\frac{2\,d-2}{2\,d-1}\Gamma_2\right)^r K_{r,2}(x)$$

```
In[5038]:= Tp = Max[2 d z[o] (Sum[((2 d - 2)/(2 d - 1) Gamma2)^r K[r, 2, {0}], {r, 1, 3}]),
    2 d z[o] (1/(2 d) + Sum[((2 d - 2)/(2 d - 1) Gamma2)^r K[r, 2, {1}], {r, 1, 3}])]
 Tp (1 + 2 Tbar) (* Condition, improved using Haras remark *)
 Tp (1 + 2 Tbar) < 1 (* Condition, improved using Haras remark *)

Out[5038]= 0.280359

Out[5039]= 0.580686

Out[5040]= True
```

Algorithm to find good values for the constants

The following is a semi-automatic procedure to find appropriate values for the constants Γ_i and c_i.

Initially, you are required to make a guess for good values for the constants and compile the file once. Good first guesses are always the bounds for the initial point/probability z_I, which are independent of the constants. Then, we deactivate the initial definition in the top of the document and compile the whole file multiple times. At each run the code will update the constants and hopefully below converges to a fixed point for the parameters.

```
In[5041]:= {d, Gamma1, Gamma2, c[1, 6, 0], c[0, 0, 1], c[1, 0, 1], c[1, 1, 1], c[1, 2, 1],
  c[1, 3, 1], cmu}
 Gamma1 = Max[boundF1[i], boundF1[o]] + 0.0000000001;
 Gamma2 = Max[boundF2[i], boundF2[o]] + 0.0000000001;

 c[1, 6, 0] = boundF3[1, o] * 1/Gamma3 + 0.000000001;
 c[0, 0, 1] = boundF3[2, o] * 1/Gamma3 + 0.000000001;
 c[1, 0, 1] = boundF3[3, o] * 1/Gamma3 + 0.00000001;
 c[1, 1, 1] = boundF3[4, o] * 1/Gamma3 + 0.00000001;
 c[1, 2, 1] = boundF3[5, o] * 1/Gamma3 + 0.00000001;
 c[1, 3, 1] = boundF3[6, o] * 1/Gamma3 + 0.00000001;
 cmu = mubOverMu[o] + 0.000001;

 {d, Gamma1, Gamma2, c[1, 6, 0], c[0, 0, 1], c[1, 0, 1], c[1, 1, 1], c[1, 2, 1],
  c[1, 3, 1], cmu}

Out[5041]= {11, 1.01306, 1.07513, 0.008714, 0.0661, 0.108, 0.05352, 0.0404, 0.022, 1.00297}

Out[5051]= {11, 1.01306, 1.07512, 0.00871352, 0.0660248,
  0.107997, 0.0535177, 0.040359, 0.0219457, 1.00297}
```